



PRESENTA

62 Programmi

per il

VIC 20

C 16

PLUS 4



**I LIBRI DI
systems**

SOMMARIO

0. Introduzione	5
1. Giochi e passatempi	7
Testa o croce	7
Tiro di dadi	8
Lancio di quattro dadi	9
Messaggio segreto	10
Beep	12
Test di memoria	13
Crea un quiz	15
Parola d'ordine multipla	17
Invertitore di parole	19
Decide il computer	20
Striscia scorrevole	22
Tre carte	23
Divisione in sillabe	25
Lui e lei	27
Sorpresa	29
Nevicata	33
Parallelepipedo	34
Piano	35
Prova schermo	38
Nim	41
Generatore di labirinti	45
Grafica	47
Bioritmi	49
2. Numeri	54
Differenza date	54
Giorno della settimana	56
Identificatore alfabetico	58
Percentuali	60

Sistemi lineari	61
Ordinamento alfabetico	64
Da esadecimale in decimale	66
Numeri dei caratteri	68
Simulatore di calcolatrice RPN	69
Invertitore di numeri	71
Integrali	72
Deviazione standard	74
Divisione di fattori	75
Funzioni logiche	76
Statistiche in base 1000	77
Numeri casuali: ordinamento maggiore/minore	78
Media numerica casuale	80
Numero medio di una serie	82
Numeri casuali: distribuzione	83
Classificazione	85
Decimale binario	88
Binario decimale	89
Equivalenze	90
Sequenze	94
Trova il numero	96
Scarta l'intero	98
Arrotondamenti	99
Arrotondamenti a doppia faccia	103
Dalla percentuale al decimale	105
3. Gestionali	107
Comparatore di introiti	107
Pubblicità: costo per unità di vendita	109
Contapezzi	110
Inventario	111
Ore e salari	113
Al penny più vicino	114
Aumento di valore monetario	116
Pagamento di automobili	118
Generatore di caratteri	120
Effetti sonori	124

INTRODUZIONE

Quando la Commodore ha introdotto sul mercato il C 16 ha voluto senza alcun dubbio una chiave alternativa al VIC 20 sia dal punto di vista del costo che delle prestazioni. Il C 16 ha senza dubbio delle peculiarità notevoli rispetto al suo predecessore, ma l'immediatezza, la facilità d'impiego e le mete che ci si prefigge di raggiungere sono le medesime.

Ecco perchè abbiamo voluto fondere in un unico volume dei programmi validi, in linea di massima, sia per l'uno che per l'altro home computer. La non totale compatibilità fra C 16 e VIC 20 ci ha obbligati a "segnare" opportunamente i programmi che vi offriamo. Ecco quindi che quei listati contrassegnati con un asterisco si riferiscono esclusivamente al computer VIC 20. I programmi invece che presentiamo con due asterischi potranno essere utilizzati indifferentemente dai possessori di VIC 20 che da quelli di C 16.

Per ultimi, non per importanza naturalmente, sono quei programmi che presentano tre asterischi. In questo caso specifico occorre che i possessori di C 16 apportino il seguente intervento di una semplicità banale:

Quando si incontra l'istruzione:

POKE 36879, un numero

occorre digitare:

COLOR 0,1

oppure quando si incontrano altri tipi di istruzioni BASIC:

POKE numero, numero

occorre ignorare, letteralmente, il comando.

CARATTERI E CODICI

Simbolo	Testi utilizzati	Denominaz.	Simbolo	Testi utilizzati	Denominaz.	Simbolo	Testi utilizzati	Denominaz.	Simbolo	Testi utilizzati	Denominaz.
	Shift Ctr/Home Clear			Ctr 1	Nero		Commodore 1 Aranc			F1	
	Ctr/Home Home			Ctr 2	Bianco		Commodore 2 Marr			Shift F1	F2
	Shift Ctrr (S) Up			Ctr 3	Rosso		Commodore 3 Rossa			F3	F3
	Ctrr (S) Down			Ctr 4	Azzur		Commodore 4 Grigio 1			Shift F3	F4
	Shift Ctrr (D) Left			Ctr 5	Viola		Commodore 5 Grigio 2			F5	F5
	Ctrr (D) Right			Ctr 6	Verde		Commodore 6 Verde 2			Shift F5	F6
	Ctr 6	Rvs		Ctr 7	Bleu		Commodore 7 Celeste			F7	F7
	Ctr 0	Rvoff		Ctr 8	Giallo		Commodore 8 Grigio 3			Shift F7	F8

* *

1. Giochi e passatempi

* *

TESTA O CROCE * *

Questo simpatico programma permette di trovare una soluzione ai dubbi con il classico sistema del "testa o croce".

Scrivi il programma.

Fallo girare.

Sarà proprio il tuo Commodore che scriverà TESTA o CROCE dopo ogni lancio.

Per un nuovo lancio, premi il tasto RETURN.

La linea 10 pulisce lo schermo.

Le linee 20 e 30 generano un numero casuale (0 oppure 1) e a seconda dell'esito, scrivono "testa" o "croce".

Le linee 50, 60 e 70 permettono l'iterazione del programma tramite il tasto RETURN.

```
10 PRINT"CLEARI":REM  PULIZIA SCHERMO
20 Q=RND(-1)
30 IF (INT(2*(RND(1))))<1 THEN PRINT"TESTA":GOTO 50
40 PRINT"CROCE"
50 PRINT:PRINT:PRINT
60 PRINT"RITENTA SARAI PIU' FORTUNA TO-!"
70 INPUT "PREMI RETURN";K$
80 CLR
90 GOTO 10
```

TIRO DI DADI * *

Ecco un semplice sistema per lanciare due dadi e mostrarne i risultati.
Le linee 20-30 trovano un numero casuale tra 1 e 6 e lo associano ad A.
Le linee 40-50 fanno lo stesso per il secondo dado (B).
Le linee 60-70 scrivono i risultati insieme ad un messaggio esplicativo.

```
10 PRINT"[CLEAR]"
20 A=INT(7*RND(1))
30 IF A<1 THEN 20
40 B=INT(7*RND(1))
50 IF B<1 THEN 40
60 PRINT"PRIMO DADO";A
70 PRINT"SECONDO DADO";B
80 FOR I=1 TO 10:PRINT:NEXT
90 PRINT"VUOI LANCIARE"CHR$(13)"ANCORA
   I DADI ?"
100 PRINT"PREMI UN TASTO"
110 GET A$:IF A$="" THEN 110
120 GOTO 10
```


LANCIO DI QUATTRO DADI * *

Due dadi non ti bastavano?

Ecco come averne quattro per ogni lancio!

Naturalmente, questo programma lavora esattamente come il precedente, tranne che il comando FOR/NEXT nelle linee 50-140 fa ripetere al computer la fase di generazione e scrittura dei lanci. Se hai bisogno di sei, otto o dieci lanci, non devi far altro che sostituire al numero due nella linea 50 il numero tre, quattro o cinque.

```
10 PRINT"[CLEAR]"
20 PRINT"LANCIA QUATTRO DADI"
30 INPUT "PREMI RETURN";K$:PRINT:PRINT
40 PRINT"[CLEAR]"
50 FOR I=1 TO 2
60 DL=INT(7*RND(1))
70 IF DL<1 THEN 60
80 DR=INT(7*RND(1))
90 IF DR<1 THEN 80
100 PRINT"*****"
110 PRINT"*";DL;"*";DR;"*"
120 PRINT"*****"
130 PRINT:PRINT
140 NEXT
150 PRINT:CLR :GOTO 20
```

MESSAGGIO SEGRETO

★ ★

I messaggi segreti possono essere molto divertenti!

Spesso sono formati da codici dove le lettere sono sostituite da numeri. In questo programma, il computer fa proprio questo: associa un numero di tre cifre ad ogni lettera dell'alfabeto.

Usando questi numeri, si possono mandare messaggi segreti assolutamente indecifrabili.

E' molto improbabile che numeri uguali siano associati a lettere diverse, dal momento che i numeri disponibili vanno da zero a 999.

Fai attenzione, quando batti il programma, di separare con virgole le lettere nella linea 30.


La linea 90 fa scrivere i risultati in due colonne.

```
10 PRINT"[CLEAR]":REM  PULIZIA SCHERMO
20 Z=RND(-TI)
30 DATA A,B,C,D,E,F,G,H,I,J,K,L,M,N,O,
   P,Q,R,S,T,U,V,W,X,Y,Z
40 FOR N=1 TO 13
50 C=INT(1000*RND(1))
60 READ L$
70 D=INT(1000*RND(1))
80 READ J$
90 PRINTL$;" ";C,J$;" ";D
100 NEXTN
110 GET K$
```

```
120 IF K$="" THEN 110
```

```
130 RESTORE
```

```
140 GOTO 10
```



BEEP ★

Puoi far suonare il tuo computer a comando.

Il comando FOR/NEXT delle linee 20 e 90 provoca la ripetizione automatica del "bip" per dieci volte.

Ovviamente puoi cambiare il numero degli squilli sostituendo il numero 10 nella linea 20.

La linea 30 fa apparire la parola "BEEP" sullo schermo.

Le linea 60 fa durare il singolo suono.

```
10 PRINT"[CLEAR]":REM  PULIZIA SCHERMO
20 FOR L=1 TO 10
30 PRINT"BEEP"
40 POKE 36878,15
50 POKE 36876,220
60 FOR T=1 TO 30:NEXTT
100 POKE 36878,0
110 FOR I=1 TO 300:NEXTI,L
```


TEST DI MEMORIA * *

Questo simpatico programma metterà a dura prova la vostra memoria !!
Il computer vi mostrerà un numero per brevi istanti dopodiché vi chiederà di ripeterlo.


Se per 3 volte sbaglierete vi verrà visualizzato un messaggio.
Diversamente vi verrà proposto un altro numero.

```
10 PRINT"[CLEAR][TF8]":CLR
20 Q=RND(-TI):Z=1
30 S=1000*RND(1)
40 N=INT(S*Z)
50 PRINT"RICORDATI--->":N
60 IF W=3 THEN PRINT"DIMENTICALO!":GOTO
   0 140
70 FOR T=1 TO 1000:NEXT
80 PRINT"[CLEAR]"
90 INPUT "CHE NUMERO ERA":S
100 IF S<>N THEN PRINT"***HAI SBAGLIATO
   ***":W=W+1:GOTO 60
110 PRINT"***ESATTO***":R=R+1:Z=Z+10
120 PRINTR;"TENTATIVO"
125 FOR T=1 TO 1000:NEXT
130 PRINT:GOTO 30
140 PRINT:PRINT"HAI AVUTO":R:PRINT"RISP
   OSTE ESATTE"
```

```

150 PRINT:PRINT"PREMI UN TASTO PER      R
    ICOMINCIARE"
170 GET A$:IF A$="" THEN 170
190 CLR :GOTO 10

```



CREA UN QUIZ * *

Uno degli usi più affascinanti per il vostro computer è impiegarlo per comunicazioni visive di carattere istruttivo, educativo o semplicemente divertente.

Un utile modo di avviare queste conversazioni può essere un quiz come quello qui presentato.

I dati del quiz (la "conoscenza" del computer) sono nelle linee 20-70.

Fai attenzione alle virgole e agli spazi che devono essere riprodotti correttamente quando trascrivi il programma.

Le righe 90 e 100 generano un numero casuale fra 1 e 11.

La riga 110 seleziona la riga nei DATA per la domanda.

Le linee 120-140 prendono l'esatto numerale dalla linea DATA selezionata.

Le linee 160-180 scrivono la domanda del quiz sullo schermo, mentre la 190 legge nei DATA la risposta esatta.

Tu fornisci la risposta che viene confrontata con quella esatta (220—240).

Naturalmente il quiz può essere ampliato.

Infatti in questo esempio, aggiungendo nuovi presidenti, in nuove linee DATA si aggiungono possibilità di gioco.

```
10 PRINT"[CLEAR]"
20 DATA  PRIMO, GEORGE WASHINGTON
30 DATA  SECONDO, JOHN ADAMS
```

```

40 DATA TERZO, THOMAS JEFFERSON
50 DATA QUARTO, JAMES MADISON
60 DATA QUINTO, JAMES MONROE
70 DATA SESTO, JOHN QUINCY ADAMS
75 PRINT"QUANTI"
80 PRINT"PRESIDENTI AMERICANI"
85 PRINT"SAPRESTI ELENCARE?"
90 R=INT(12*RND(1))
100 IF R<1 THEN 90
110 IF INT(R/2)=R/2 THEN R=R+1
120 FOR L=1 TO R
130 READ S$
140 NEXTL
150 PRINT:PRINT
160 PRINT"CHI FU IL "
170 PRINTS$;" PRESIDENTE"
180 PRINT"DEGLI STATI UNITI?"
190 READ C$
200 INPUT D$
210 FOR L=1 TO 3:PRINT:NEXTL
220 IF D$=C$ THEN PRINT"ESATTO!":GOTO 2
    40
230 PRINT"SBAGLIATO!"
240 PRINT"IL ";S$
250 PRINT"PRESIDENTE FU"
260 PRINTC$
270 RESTORE
280 PRINT:PRINT
290 GOTO 90

```


PAROLA D'ORDINE MULTIPLA * *

Questo è un sofisticato sistema di parole d'ordine che permette l'accesso al programma ad alcune persone, ognuna delle quali avrà un personale numero di codice ed una corrispondente parola d'ordine. Ogni utilizzatore deve immettere il suo codice e quindi la sua parola d'ordine.


Se il numero è sbagliato la parola non potrà mai essere esatta.

Se invece il numero è giusto ma la parola non corrisponde, niente da fare egualmente!

Puoi aggiungere altri utenti autorizzati a questo programma scrivendo nuove linee nella subroutine posta fra le righe 300 e 340.

```
10 PRINT"ICLEAR!"
20 PRINT"DIGITA IL TUO NUMERO PERSONA
   LE":INPUT UA
30 GOSUB 300
50 PRINT"PAROLA D'ORDINE":INPUT P$
60 IF P$="" THEN PRINT"[2 UP]":GOTO 50
70 IF P$=PW$ THEN 110
90 PRINT"DATI NON ACCETTABILI"
100 END
110 PRINT"OK! IL CODICE E'          ESATTO"
```

```
120 PRINT"PUOI ACCEDERE AL      PROGRAM  
    MA PROTETTO"  
130 END  
300 IF UA=12345 THEN PW$="ZEBRA"  
310 IF UA=23456 THEN PW$="COBRA"  
320 IF UA=34567 THEN PW$="TIGRE"  
330 IF UA=45678 THEN PW$="CONDOR"  
340 RETURN
```



INVERTITORE DI PAROLE

★ ★

Digita una parola qualsiasi od un numero e scoprilo scritto alla rovescia sullo schermo.

Alla fine il computer attende che tu prema un qualsiasi tasto per ricominciare.

```
10 PRINT"[CLEAR]"
20 PRINT"PAROLA/NUMERO: ":INPUT N$
30 IF N$="" THEN PRINT"[2 UP]":GOTO 20
35 PRINT"[CLEAR]"
40 L=LEN(N$)
45 PRINT"ORIGINALE:":PRINTN$
50 PRINT"ROVESCiato:"
60 FOR Y=L TO 1 STEP -1
70 B$=MID$(N$,Y,1)
80 PRINTB$;
90 NEXTY
100 FOR K=1 TO 10:PRINT:NEXT
110 PRINT"PREMI UN TASTO"
120 PRINT"PER CONTINUARE"
130 GET A$
140 IF A$="" THEN 130
150 B$=""
160 GOTO 10
```

DECIDE IL COMPUTER

★ ★

Ti ricordi il programma che diede tanta popolarità al computer?

Ora la potenza del calcolatore rende possibile una versione migliorata di quel programma.


In questa edizione è possibile una scelta tra otto alternative.

Dopo ogni esecuzione il computer attende che tu prema un qualsiasi tasto per ricominciare.

```
10 DATA LASCIA PERDERE
20 DATA ACCETTA L'AFFARE
30 DATA SI!
40 DATA FORSE...
50 DATA E' MEGLIO CHE RIORDINILE TUE
   IDEE
60 DATA SOPRASSIEDI
70 DATA NO!!!
80 DATA OGGI NON USCIRE DI      CASA
90 Q=RND(-TI)
100 PRINT"[CLEAR]"
110 N=INT(9*RND(1))
120 IF N<1 THEN 110
130 FOR L=1 TO N
140 READ A$
150 NEXT L
160 PRINTA$
```



```
170 GET K$.  
180 IF K$="" THEN 170  
190 RESTORE  
200 GOTO 100
```



STRISCIA SCORREVOLE

★ ★ ★

Inserite nel vostro calcolatore una frase per ogni richiesta di input (il numero totale lo stabilirete voi) e la vedrete scorrere da destra a sinistra al centro del vostro televisore come avviene per le pubblicità luminose di molti negozi.

```
10 POKE 36879,8:PRINTCHR$(5)
20 PRINTCHR$(147)
30 PRINT"QUANTE LINEE DI INPUT VUOI":I
   INPUT E
40 DIM A$(E):FOR K=1 TO E
50 INPUT ">":A$(K):A$(K)="
   "+A$(K):NEXTK
100 PRINTCHR$(147)
110 FOR P=1 TO E
120 FOR I=22 TO 22+LEN(A$(P))
130 PRINTCHR$(19):PRINT"[10 DOWN]"; TAB
   (22):MID$(A$(P),I-21,21);" "
140 FOR J=1 TO 100:NEXTJ,I
150 NEXTP
160 GET A$:IF A$="" THEN 110
200 END
```

TRE CARTE * *

Un antico gioco arabo basato sulla velocità e sull'abilità del prestidigiante consisteva nel mescolare tre carte disposte a dorso in su sopra un tavolo e di far scommettere a qualche sprovveduto passante quale fosse l'asso di cuori.

Vi viene qui riproposto lo stesso gioco avvincente, nel quale però il vostro avversario non sarà la mano veloce di un arabo ma la sorte che casualmente assegnerà la posizione dall'asso di cuori.

```
1 P=0
5 Z$="?":X$="?":Y$="?"
10 GOSUB 1000:GOSUB 1030
20 F=RND(1)*6+1
25 IF F>6 THEN F%=6
30 IF F>0 AND F<2 THEN 35
31 IF F>2 AND F<4 THEN 36
32 IF F>4 AND F<6 THEN 37
35 Z$="♠":Y$="♣":X$="♥":C=1:GOTO 40
36 Z$="♣":Y$="♠":X$="♥":C=3:GOTO 40
37 Z$="♥":Y$="♠":X$="♣":C=2:GOTO 40
40 GOSUB 1000
45 FOR G=1 TO 2000:NEXT G
50 PRINT"[CLEAR][3 DOWN][2 RIGHT]HAI P
UNTATO SU":J
55 IF C=J THEN 70
```

```

60 IF C<>J THEN 80
70 PRINT"[3 DOWN][3 RIGHT]HAI VINTO!":
  P=P+10000:GOTO 90
80 PRINT"[3 DOWN][3 RIGHT]HAI PERSO!":
  P=P-10000
90 PRINT"[DOWN]SEI A QUOTA È"P
100 INPUT "[3 DOWN][3 RIGHT]ANCORA";E$
110 IF LEFT$(E$,1)<>"S" THEN 100
117 PRINT"[3 DOWN][3 RIGHT][RVS]< PREMI
  UN TASTO >[RVOFF]"
120 GET W$:IF W$="" THEN 120
130 GOTO 5
140 END
175 PRINT"[DOWN]SEI A QUOTA "P" LIRE"
1000 PRINT"[CLEAR][3 DOWN]"; TAB(3)"[RVS
  ][RVOFF]"; TAB(10)"[RVS]2[RVOFF]";
  TAB(17)"[RVS]3[RVOFF]"
1010 PRINT"[DOWN][RIGHT]_____[2 RIGHT]____
  _____[2 RIGHT]_____"
1020 S$="[RIGHT] | | | | "
1021 FOR I=1 TO 5:PRINTS$:NEXTI
1025 PRINT" | "Z$" | | "Y$" | | "X$"
  | "
1026 FOR I=1 TO 5:PRINTS$:NEXT
1027 PRINT"[RIGHT]_____[2 RIGHT]_____[2
  RIGHT]_____"
1029 RETURN
1030 PRINT"[DOWN]DOVE È L'ASSO DI C
  UORI"
1040 INPUT J
1045 IF J<1 OR J>3 THEN PRINT"[2 UP]":GO
  TO 1040
1050 RETURN

```

DIVISIONE IN SILLABE

★ ★

Questo utile programmino dividerà in sillabe qualsiasi parola voi digiterete.

```
200 PRINT"[CLEAR]"
210 S=1:PRINT
220 PRINT"VOCABOLO":INPUT A$:PRINT
230 IF A$="" THEN END
240 IF SOLEN(A$) THEN PRINT"[LEFT]":GOTO 210
250 X$=MID$(A$,S,1):GOSUB 460
260 IF A=0 THEN 430
270 X$=MID$(A$,S+1,1):GOSUB 460
280 IF A=0 THEN 360
290 IF X$="I" THEN 320
300 IF MID$(A$,S,1)="I" OR MID$(A$,S,1)="U" THEN 430
310 GOTO 440
320 X$=MID$(A$,S+2,1):GOSUB 460
330 IF S>1 THEN IF MID$(A$,S-1,2)="OU" AND A=-1 THEN PRINTMID$(A$,S,2):S=S+2:GOTO 240
340 IF A=-1 THEN 440
350 GOTO 430
360 IF S+2>LEN(A$) THEN 450
```

```

370 X$=MID$(A$,S+2,1):GOSUB 460
380 IF A=-1 THEN 440
390 IF MID$(A$,S+1,1)=MID$(A$,S+2,1) TH
   EN 450
400 A1$=MID$(A$,S+1,1):IF A1$="S" OR A1
   $="G" THEN 440
410 A2$=MID$(A$,S+2,1):IF A2$="R" OR A2
   $="L" OR A2$="H" THEN 440
420 GOTO 450
430 PRINTMID$(A$,S,1):S=S+1:GOTO 240
440 PRINTMID$(A$,S,1);"-":S=S+1:GOTO 2
   40
450 PRINTMID$(A$,S,2);"-":S=S+2:GOTO 2
   40
460 A=X$="A" OR X$="E" OR X$="I" OR X$=
   "O" OR X$="U":RETURN

```

LUI E LEI * *

Questo divertente gioco di società usava solitamente come supporto dei fogli di carta su cui ogni giocatore scriveva segretamente la propria risposta, e la continua rotazione di foglietti da un giocatore al vicino ad ogni nuova domanda, provocava una combinazione di esilaranti "storielle".

Questa volta il supporto vi verrà fornito dalla modesta memoria del vostro VIC—20.

```
10 S$(1)="[2 DOWN][RVS]CHI E' LUI"
20 S$(2)="[2 DOWN][RVS]CHI E' LEI"
30 S$(3)="[2 DOWN][RVS]DOVE SONO"
40 S$(4)="[2 DOWN][RVS]COSA STANNO FAC
ENDO"
50 S$(5)="[2 DOWN][RVS]COSA DICE LUI A
LEI"
60 S$(6)="[2 DOWN][RVS]COSA DICE LEI A
LUI"
70 S$(7)="[2 DOWN][RVS]COSA DICE LA GE
NTE DI LORO"
80 S$(8)="[2 DOWN][RVS]COSA FACEVANO P
RIMA DICONOSCERSI"
90 S$(9)="[2 DOWN][RVS]CHE LAVORO FA L
UI"
100 S$(10)="[2 DOWN][RVS]CHE ATTIVITA'
```

```

        SVOLGE LEI"
110 PRINT"[CLEAR][5 DOWN]QUANTI GIOCATO
    RI      (MAX.10)":INPUT G
120 IF G<1 OR G>10 THEN 110
130 FOR W=1 TO G
140 FOR T=1 TO 10
150 PRINT"[CLEAR]"S$(T)
160 INPUT "[DOWN]";A$(T,W)
170 NEXTT
180 IF W<>G THEN PRINT"[CLEAR][2 DOWN]H
    AI FINITO!":PRINT"[DOWN]TOCCA AL PR
    OSSIMO GIOCATORE!":GOTO 200
190 PRINT"[CLEAR][DOWN][RVS]OK![2 DOWN]
    ATTENDETE!":FOR I=1 TO 1000:NEXT
200 FOR I=1 TO 1000:NEXTI:NEXTW
210 D=INT(G/2):PRINT"[CLEAR]"
220 FOR K=1 TO 10
230 PRINTS$(K):PRINT"[DOWN]"A$(K,D)
240 D=D+1:IF D=G+1 THEN D=1
250 FOR I=1 TO 1000:NEXT
260 NEXTK
270 IF D=INT(G/2) THEN END
280 PRINT"***PREMI UN TASTO***"
290 GET Z$:IF Z$="" THEN 290
300 GOTO 220

```


SORPRESA *

Ecco un classico gioco spaziale per il vostro computer VIC—20. Dovete cercare di evitare le collisioni con gli elementi alieni e gli asteroidi. I tasti M e Z vi permetteranno di muovervi rispettivamente verso destra e verso sinistra.

```
150 POKE 650,128
200 PRINT"[CLEAR][2 DOWN][RVS]COMANDI[R
V OFF]"
250 PRINT"[2 DOWN][RVS]Z[R V OFF] SINISTR
A"
300 PRINT"[2 DOWN][RVS]M[R V OFF] A DESTRA"
350 PRINT"[2 DOWN][RVS]*[R V OFF] ALIENI"
400 PRINT"[2 DOWN][RVS]●[R V OFF] ASTEROI
DI"
450 INPUT "[2 DOWN][RVS]IL TUO NOME?[R V
OFF]";P$
500 PRINT"[CLEAR]"
550 PRINT"[2 DOWN][RVS]IL TUO NOME E'[R
V OFF]";P$
600 FOR H=1 TO 1000:NEXT
650 PRINT"[CLEAR]"
700 POKE 36879,8
750 H=0:S=0:Z=0
```

```

800 Q=50
850 A=7900:B=11:D=8164
900 FOR J=1 TO 24:PRINT:NEXT
950 FOR K=1 TO 600
1000 POKE D+INT(32*RND(1)),46
1050 R=INT(25*RND(1))
1100 IF R=1 THEN 3450
1150 FOR J=1 TO Q:NEXT
1200 Q=Q-0.25
1250 POKE A+B,32
1300 PRINT
1350 GET A$
1400 IF A$="" THEN 1650
1450 IF B=1 AND A$="M" THEN H=H+1
1500 IF B=21 AND A$="Z" THEN H=H+1
1550 IF B>1 AND A$="Z" THEN B=B-1
1600 IF B<21 AND A$="M" THEN B=B+1
1650 POKE A+B,22
1700 IF PEEK((A+B)+22)=46 THEN 1900
1750 IF PEEK((A+B)+22)=42 THEN 3700
1800 NEXTK
1850 GOTO 2600
1900 POKE 36879,42
1950 PRINT"[CLEAR]"
2000 PRINT"[7 DOWN][BIANCO]ASTRONAVE DIS
      TRUTTA!!"
2050 POKE 36877,220
2100 S=S+1
2150 FOR L=15 TO 0 STEP -1
2200 POKE 36878,L
2250 FOR M=1 TO 100
2300 NEXTM
2350 NEXTL
2400 POKE 36877,0

```

```

2450 POKE 36878,0
2500 POKE 36879,8
2550 GOTO 1000
2600 PRINT"[CLEAR]":FOR K=1 TO 1000:NEXT
      :POKE 36879,106
2650 FOR R=1 TO 5:PRINT:NEXTR
2700 Z=(H*50)-(S*100)+(X*200)
2750 PRINT"[CLEAR]PUNTEGGIO DI ";P$
2800 PRINT"[DOWN][RVS]TOTALE PUNTI[RVOFF
      ] ";Z
2850 PRINT"[DOWN]ASTERIODI INVESTITI";S
2900 PRINT"[DOWN][RVS]ALIENI Distrutti "
      ;X
2950 PRINT"[DOWN]GIOCHI ANCORA?"
3000 INPUT "[DOWN][RVS]S/N[RVOFF]";W$
3050 IF W$="S" THEN 500
3100 POKE 36879,44
3150 PRINT"[CLEAR]"
3200 FOR U=1 TO 11:PRINT:NEXT
3250 PRINT"[CLEAR][9 DOWN][2 RIGHT][GIAL
      LO]OKEY"
3300 PRINT
3350 PRINT"[RVS][3 DOWN][GIALLO]#####
      #CIAO#####"
3400 END
3450 FOR P=1 TO 10
3500 POKE D+INT(32*RND(1)),42
3550 FOR J=1 TO Q:NEXTJ
3600 POKE A+B,32
3650 GOTO 1300
3700 POKE 36879,127
3750 PRINT"[CLEAR]"
3800 PRINT"[4 DOWN][NERO]#####ALIENO COLP
      ITO#####"

```

```
3850 POKE 36878,15
3900 FOR L=1 TO 15
3950 POKE 36876,160
4000 FOR M=1 TO 10:NEXTM
4050 POKE 36876,0
4100 FOR M=1 TO 10:NEXTM
4150 NEXTL
4200 POKE 36878,0
4250 X=X+1
4300 POKE 36879,8
4350 GOTO 1000
```

NEVICATA *

Bianchi fiocchi di neve compaiono sul vostro schermo fino a che non premerete il tasto asterisco *.

Non sarà molto utile, ma è senz'altro piacevole da guardare.

```
10 PRINT"[CLEAR]"
20 SL=INT((7680+22*23)*RND(1))
30 IF SL<7680 THEN 20
40 POKE SL,42:POKE SL-7680+38400,INT(7
  *RND(1))
50 GET K$
60 IF K$="*" THEN 80
70 GOTO 20
80 PRINT"[CLEAR]"
90 PRINT"PREMERE (Q) PER          FINIRE"
110 PRINT"PREMERE (R) PER          RICOMIN
    CIARE"
120 GET HY$
130 IF HY$="" THEN 120
140 IF HY$="Q" THEN 170
150 IF HY$="R" THEN 10
160 GOTO 120
170 END
```

PARALLELEPIEDO

★ ★

Questa sezione del libro è dedicata alla grafica a colori.

Troverete un certo numero di nuove ed interessanti applicazioni delle capacità grafiche del vostro computer Commodore.

Gli esempi proposti possono essere modificati o combinati tra loro per adattarsi a qualsiasi esigenza.

I titoli rappresentano solo l'idea originale da cui sono sorti i diversi programmi, ma la vostra fantasia potrà dare vita a nuove ed infinite creazioni.

Possono essere modificati i colori degli oggetti e le posizioni sullo schermo, nonché i vari movimenti.

```
10 PRINT"[CLEAR]":REM  PULIZIA SCHERMO
20 PRINT:PRINT:PRINT
30 PRINTSPC(5);CHR$(111);
40 FOR L=1 TO 10:PRINTCHR$(163):NEXT
50 PRINTCHR$(112)
60 FOR L=1 TO 13
70 PRINTSPC(5);CHR$(165);SPC(10);CHR$(
  167)
80 NEXTL
90 PRINTSPC(5);CHR$(108);
100 FOR L=1 TO 10:PRINTCHR$(164):NEXT
110 PRINTCHR$(186)
```

PIANO *

Trasformate il vostro Computer in un formidabile pianoforte e lasciate la vostra fantasia libera di spaziare tra le note musicali per creare tanti piacevoli motivi.

```
1 PRINT"[CLEAR][NERO]";
2 PRINT"[RVS]      SINTETIZZATORE      [R
  VOFF]"
4 PRINT"[ROSSO]TIMBRI F#1 - F#7[BIANC
  O]:"
7 PRINT"[NERO]CRSR ← NOTA SOMMATA  C
  ONTINUA":PRINT:PRINT"CRSR ↑ NOTA SO
  MMATA  NON CONTINUA"
8 X=36876:V=15:P=15
9 PRINT:PRINT
10 Z=PEEK(197)
12 IF Z=64 THEN POKE X,0:POKE 36878,P:
  GOTO 10
18 POKE 36878,V:IF V<1 THEN V=1
19 IF V>15 THEN V=15
20 IF Z>22 AND Z<44 THEN A=Z-22:GOTO
  35
25 IF Z>44 THEN B=Z-44:GOTO 40
30 ON ZGOTO 52,54,56,58,60,62,64,10,66
  ,68,70,72,74,76,10,10,78,80,82,84,8
```

6,88

35 ON AGOTO 106,10,10,91,93,95,97,99,1
11,110,90,92,94,96,98,10,102,10,79,
81,83,85

40 ON BGOTO 87,89,103,65,67,69,71,73,7
5,77,104,51,53,55,57,59,61,63,105

51 POKE X,135:GOTO 10

52 POKE X,143:GOTO 10

53 POKE X,147:GOTO 10

54 POKE X,151:GOTO 10

55 POKE X,158:GOTO 10

56 POKE X,159:GOTO 10

57 POKE X,163:GOTO 10

58 POKE X,167:GOTO 10

59 POKE X,175:GOTO 10

60 POKE X,179:GOTO 10

61 POKE X,183:GOTO 10

62 POKE X,187:GOTO 10

63 POKE X,189:GOTO 10

64 POKE X,191:GOTO 10

65 POKE X,201:GOTO 10

66 POKE X,203:GOTO 10

67 POKE X,207:GOTO 10

68 POKE X,209:GOTO 10

69 POKE X,212:GOTO 10

70 POKE X,215:GOTO 10

71 POKE X,217:GOTO 10

72 POKE X,219:GOTO 10

73 POKE X,221:GOTO 10

74 POKE X,223:GOTO 10

75 POKE X,225:GOTO 10

76 POKE X,227:GOTO 10

77 POKE X,228:GOTO 10

78 POKE X,229:GOTO 10


```

79 POKE X,231:GOTO 10
80 POKE X,232:GOTO 10
81 POKE X,233:GOTO 10
82 POKE X,235:GOTO 10
83 POKE X,236:GOTO 10
84 POKE X,237:GOTO 10
85 POKE X,238:GOTO 10
86 POKE X,239:GOTO 10
87 POKE X,240:GOTO 10
88 POKE X,241:GOTO 10
89 POKE X,243:GOTO 10
90 POKE X,244:GOTO 10
91 POKE X,245:GOTO 10
92 POKE X,246:GOTO 10
93 POKE X,247:GOTO 10
94 POKE X,248:GOTO 10
95 POKE X,249:GOTO 10
96 POKE X,250:GOTO 10
97 POKE X,251:GOTO 10
98 POKE X,252:GOTO 10
99 POKE X,253:GOTO 10
101 POKE X,254:GOTO 10
102 X=36874:GOTO 10
103 X=36875:GOTO 10
104 X=36876:GOTO 10
105 X=36877:GOTO 10
106 P=15:GOTO 10
110 FOR I=1 TO 4:POKE 36878+I,0:NEXT I:G
    OTO 10
111 P=0:GOTO 10

```

PROVA SCHERMO

★ ★ ★

Questo programmino di utilità vi consente di vedere tutte le combinazioni di colori sfondo/bordo che il vostro computer può offrirvi.

```
10 POKE 36879,14:POKE 36869,242:DIM A$(17)
20 COL$="[BIANCO][ROSSO][AZZUR][VIOLA][VERDE][BLEU][GIALLO][VIOLA][BIANCO][ROSSO][AZZUR][VIOLA][VERDE][BLEU][GIALLO][VERDE]":C=1:R$="[RVS] [RV OFF]":RI$="[HOME][23 DOWN]"
30 AS$(1)="[VERDE] ■ [ROSSO][RVS] [RV OFF]":AS$(2)="[VERDE] ■ [ROSSO][RVS] [RV OFF]":AS$(3)="[ROSSO][RVS] ■ [RV OFF]":AS$(4)="[ROSSO][RVS] ■ [RV OFF]"
40 AD$(1)="[RVS][ROSSO] [RV OFF][VERDE] ■ ":AD$(2)="[RVS][ROSSO] [RV OFF][VERDE] ■ ":AD$(3)="[RVS][ROSSO] [RV OFF] ■ ":AD$(4)="[RVS][ROSSO] [RV OFF] ■ "
50 BS$(1)="[ROSSO][RVS] ■ [RV OFF]":BS$(2)="[ROSSO][RVS] ■ [RV OFF]":BS$(3)="[VERDE] ■ [ROSSO][RVS] [RV OFF]":BS$(4)="[VERDE] ■ [ROSSO][RVS] [RV OFF]"
```

```

OFF]"
60 BD$(1)="[RVS][ROSSO] [RVOFF]" : BD
$(2)="[RVS][ROSSO] [RVOFF]" : BD$(
3)="[RVS][ROSSO] [RVOFF][VERDE] "
:BD$(4)="[RVS][ROSSO] [RVOFF][VERD
E] "
70 KH$(0)="_":KH$(1)="I [RVS]"
[RVOFF] "I":KH$(2)="I [RVS] [RVOFF]
[RVS] [RVOFF]I"
80 KH$(3)="I [RVS] [RVOFF] I":KH$(4
)="I [RVS] [RVOFF] [RVS] [RVOFF]I
":KH$(5)="I [RVS] [RVOFF] I"
90 KH$(6)="_":KH$(7)="[LEFT] V
IC 20 ":KH$(8)="COMPUTER[DOWN][9 LE
FT] VIC 20 "
100 CL$="[RVS][ROSSO] [AZZUR] [VIOL
A] [VERDE] [BLEU] [GIALLO]
[RVOFF]":NE$="[BIANCO]███"
110 PRINT"[CLEAR]":FOR I=1 TO 4:PRINT T
AB(1)AS$(I) TAB(17)AD$(I):NEXT
120 PRINTLEFT$(RI$,19):FOR I=1 TO 4:PR
INT TAB(1)BS$(I) TAB(17)BD$(I):NEXT
130 PRINTLEFT$(RI$,4):FOR I=0 TO 8:PRI
NT TAB(7)"[NERO]"KH$(I):NEXT
140 PRINTLEFT$(RI$,14):FOR I=1 TO 3:PR
INT TAB(2)CL$:NEXT
150 PRINTLEFT$(RI$,23) TAB(4)"[BIANCO]P
REMI UN TASTO":
160 PRINTLEFT$(RI$,17):FOR I=1 TO 5:PR
INT TAB(7)NE$:NEXT
170 V$=MID$(COL$,C,1):RV$=V$+R$:C=C+1:P
RINTLEFT$(RI$,7) " ":IF C=8 THEN C=
1
180 FOR I=1 TO 6:PRINTRV$:NEXT:FOR I=1

```

```

    TO 4:PRINT"[UP][LEFT]"RV$):NEXT
190 FOR I=1 TO 9:PRINTRV$):NEXT:FOR I=1
    TO 4:PRINT"[DOWN][LEFT]"RV$):NEXT
200 FOR I=1 TO 5:PRINTRV$):NEXT:FOR I=1
    TO 10:PRINT"[DOWN][LEFT]"RV$):NEXT
210 FOR I=1 TO 5:PRINT"[2 LEFT]"RV$):NE
    XT:FOR I=1 TO 5:PRINT"[DOWN][LEFT]"
    RV$):NEXT
220 FOR I=1 TO 9:PRINT"[2 LEFT]"RV$):NE
    XT:FOR I=1 TO 5:PRINT"[UP][LEFT]"RV
    $):NEXT
230 FOR I=1 TO 5:PRINT"[2 LEFT]"RV$):NE
    XT:FOR I=1 TO 9:PRINT"[UP][LEFT]"RV
    $):NEXT
240 PRINTLEFT$(RI$,7):FOR I=1 TO 6
250 PRINT"[2 RIGHT]"MID$(COL$,C+7,1)RIG
    HT$(NE$,4) TAB(16)MID$(COL$,C+6,1)R
    IGH$$(NE$,4)
260 NEXT:PRINTLEFT$(RI$,4):FOR I=0 TO
    8:PRINT TAB(7)MID$(COL$,C+5,1)KH$(I
    ):NEXT
270 POKE 36878,8:POKE 36876,INT(RND(1)*
    128)+128
280 GET X$:IF X$="" THEN 170
290 POKE 36876,0:POKE 36878,0:POKE 3687
    9,27:PRINT"[NERO][CLEAR]";

```

NIM * * *

Questo antico gioco cinese viene qui riproposto fedelmente, fornendovi perciò un avversario praticamente imbattibile.

A turno dovrete togliere da una sola fila per volta quanti fiammiferi vorrete.

Vince chi lascia all'avversario l'ultimo fiammifero da togliere.

```
20 CLR :PRINT"[CLEAR][BIANCO]":POKE 36
   879,8
30 PRINT TAB(4)"GIOCO DEL NIM"
40 FOR K=1 TO 22:PRINT"-";:NEXTK
50 PRINT"DISPORREMO DEI          FIAMMIF
   ERI SU 3 RIGHE."
70 PRINT"CIASCUNO DI NOI POTRA' TOGLIER
   E"
80 PRINT"QUANTI FIAMMIFERI          VUOLE,M
   A DA UNA"
90 PRINT"SOLA RIGA . VINCERA' CHI TOG
   LIERA'"
95 PRINT"L'ULTIMO FIAMMIFERO."
100 FOR K=1 TO 22:PRINT"-";:NEXTK
105 FOR K=1 TO 3
110 PRINT"QUANTI FIAMMIFERI          SULLA"K
    "' RIGA? ";
120 GET A$(K):IF A$(K)="" THEN 120
```

```

130 A(K)=VAL(A$(K))
140 IF A(K)<=0 THEN 120
145 PRINTA$(K):PRINT:NEXTK
155 GOSUB 1000
160 PRINT"OK - VUOI INIZIARE PERPRIMO ?
    ";
170 GET R$:IF R$="" THEN 170
175 PRINTR$
180 IF R$="S" THEN 250
190 IF R$="N" THEN 210
200 GOTO 170
210 REM MUOVE IL COMPUTER
220 PRINT"POSSO MUOVERE ? ";
230 GET R$:IF R$<>"S" THEN 230
235 PRINTR$
240 GOSUB 3000
250 REM MUOVE IL GIOCATORE
260 PRINT"VUOI TOGLIERE DALLA RIGA 1,
    2 0 3 ?";
270 GET R$:IF R$="" THEN 270
275 R=VAL(R$)
290 IF R<1 OR R>3 THEN 270
295 PRINTR$
310 PRINT"QUANTI FIAMMIFERI ?"
320 GET R$:IF R$="" THEN 320
322 IF VAL(R$)=0 THEN 320
325 PRINTR$
330 IF A(R)-VAL(R$)<0 THEN GOSUB 1000:G
    OTO 260
340 A(R)=A(R)-VAL(R$)
350 GOSUB 4000
360 IF M=0 THEN PRINT"HAI VINTO TU !":G
    OTO 500
370 GOSUB 1000

```

```

380 GOTO 210
500 REM  RICHIESTA DI PROSEGUIMENTO
510 PRINT"VUOI GIOCARE ANCORA ?"
520 GET R$:IF R$="" THEN 520
530 IF R$="S" THEN 20
540 PRINT"CIAO !":END
1000 REM  STAMPA TABELLONE
1010 PRINTCHR$(147)
1015 FOR K=1 TO 3
1017 PRINTK"/RIGA";
1020 IF A(K)=0GOTO 1040
1030 FOR J=1 TO A(K)
1035 PRINT TAB(6)"I ";
1038 NEXTJ
1040 PRINT:PRINT:NEXTK
1050 FOR K=1 TO 22
1055 PRINT"-";
1058 NEXTK:RETURN
2000 REM  ANALISI SICUREZZA DELLA MOSSA
2010 F=0:FOR K=1 TO 3:FOR J=0 TO B(K)
2015 READ B$(K):NEXTJ:RESTORE :NEXTK
2020 FOR K=1 TO 4:S=0
2030 BB$(K)="0"+MID$(B$(1),K,1)+MID$(B$(
2),K,1)
2035 BB$(K)=BB$(K)+MID$(B$(3),K,1)
2040 READ C$
2050 IF BB$(K)=C$ THEN RESTORE :GOTO 237
0
2060 GOTO 2040
2070 FOR J=1 TO 4:S(J)=VAL(MID$(C$,J,1))
2075 S=S+S(J):NEXTJ
2080 IF S=0 OR S=2 OR S=4 OR S=6 THEN 20
90
2085 IF S=8 OR S=10 OR S=12 THEN 2090

```

```

2088 F=1:RETURN
2090 NEXTK:F=0:RETURN
3000 REM DECISIONE E MOSSA
3010 FOR K=1 TO 3:B(K)=A(K):NEXTK
3020 Y=1
3030 IF B(Y)=0 THEN B(Y)=A(Y):GOTO 3050
3040 B(Y)=B(Y)-1:GOSUB 2000
3042 IF F<>0 THEN 3030
3045 FOR K=1 TO 3:A(K)=B(K):NEXTK
3048 GOTO 3110
3050 IF Y=3 THEN 3070
3060 Y=Y+1:GOTO 3030
3070 GOSUB 4000
3090 X=INT(M*RND(1)+1)
3100 A(W)=M-X
3110 GOSUB 1000
3115 GOSUB 4000
3120 IF M<>0 THEN RETURN
3130 PRINT"HO VINTO IO!":GOTO 500
4000 REM RICERCA NUM. MAX PER RIGA
4005 M=0:W=0
4010 FOR H=1 TO 3
4015 IF A(H)>M THEN M=A(H):W=H
4020 NEXTH
4030 RETURN
4050 REM :NUM MAX=M,SULLA RIGA W
5000 DATA 0000,0001,0010,0011,0100,0101,
0110,0111
5010 DATA 1000,1001,1010,1011,1100,1101,
1110,1111
31130 PRINT"HO VINTO IO !":GOTO 500

```


GENERATORE DI LABIRINTI *

Il computer riempirà il vostro schermo con un labirinto generato da valori casuali...

Lo studio di questo programma potrà esservi utile per la futura creazione di giochi a percorso.

```
95 POKE 36879,8:PRINTCHR$(5);CHR$(8);C  
   HR$(142);CHR$(147)  
100 DIM A(3)  
  
110 A(0)=2:A(1)=-44:A(2)=-2:A(3)=44  
120 WL=160:HL=32:SC=7680:A=SC+45  
140 FOR I=1 TO 21  
150 PRINTCHR$(18);  
153 FOR W=1 TO 21  
155 PRINTCHR$(32);  
157 NEXTW:PRINT  
160 NEXTI  
210 POKE A,4  
220 J=INT(RND(1)*4):X=J  
230 B=A+A(J):IF PEEK(B)=WL THEN POKE B,  
   J:POKE A+A(J)/2,HL:A=B:GOTO 220  
240 J=(J+1)*-(J<3):IF J<>X THEN 230
```

```

250 J=PEEK(A):POKE A,HL:IF J<4 THEN A=A
-A(J):GOTO 220
310 GET C$:IF C$="" THEN 310
1010 POKE A,81:J=2
1020 B=A+A(J)/2:IF PEEK(B)=HL THEN POKE
B,81:POKE A,HL:A=B:J=(J+2)+4*(J>1)
1030 J=(J-1)-4*(J=0):GOTO 1020

```



GRAFICA *

Quello che segue è all'apparenza un programma abbastanza complesso, ma che illustra brillantemente alcune delle più elementari capacità grafiche del vostro computer.



Date i valori alle variabili e, dopo una breve attesa verrà disegnata una sinusoide, e in seguito... ai più bravi scoprirlo!

```
0 C=5:B=130:D=-50:E=0
1 PRINTCHR$(147);CHR$(5):POKE 36879,8
2 FOR I=6143 TO 7679:POKE I,0:NEXTI:POKE 51,255:POKE 52,24:POKE 55,255:POKE 56,24
5 PRINT"[CLEAR]UNITA' X ";C:INPUT "[6 LEFT]";C
6 PRINT"[CLEAR]UNITA' Y ";B:INPUT "[6 LEFT]";B
7 PRINT"[CLEAR]VETTORE Y ";D:INPUT "[6 LEFT]";D
8 PRINT"[CLEAR]VETTORE X ";E:INPUT "[6 LEFT]";E
9 PRINTCHR$(147)
20 POKE 36869,254:H=7933:G=6144:X=-89
30 DEF FNF(X)=SIN(X)/X
40 X=X+1:IF X>87 THEN 75
45 IF X/C-E=0 THEN 40
```

```

50 Y=INT(B*FNFX/C-E)+D:IF Y>95 OR Y<
    -88 THEN 40
55 MZ=X/8:Z=X-MZ*8:NZ=Y/8:T=Y-NZ*8
60 A=H-22*NZ+MZ:IF PEEK(A)>32 THEN L=
    PEEK(A):GOTO 70
65 J=J+1:L=J:IF J=32 OR J=31 THEN 65
70 K=G+(7-T)+8*L:POKE K,2+(7-Z) OR PEE
    K(K):POKE A,L:GOTO 40
75 GET A$:IF A$="" THEN 75
80 PRINTCHR$(147):POKE 36869,240:J=0:L=
    0:GOTO 2

```



BIORITMI *

Un utile programma per conoscere il vostro attuale stato di salute fisico, psicologico e mentale.

I bioritmi sono tre cicli sinusoidali rispettivamente di 23, 28 e 33 giorni riguardanti il primo lo stato fisico di un individuo, il secondo lo stato emotivo e l'ultimo quello intellettuale.

Ogni volta che una curva attraversa l'asse delle ascisse dall'alto verso il basso si ha un calo della propria potenzialità.

Il programma prevede inoltre il confronto della compatibilità di diversi bioritmi.

```
1 PRINT"[CLEAR]"
100 DIM A(12),X(3),J(3),CD(3,2),R(3),S(
3):FOR I=1 TO 12:READ A(I):NEXT:DEF
FNY(I)=28-(I/4=INT(I/4))
102 P(1)=16:P(2)=5:P(3)=9:M$="GENFEBMAR
APRMAGGIULUGAGOSETOTTNOVDIC":W$="LM
MGVSD"
108 SP$=" " F$="[HOM
E][22 DOWN]"
110 DEF FNP(I)=8119-INT(I+11.5)*22+INT(
(Q+1)/2)
120 DEF FNS(I)=8*SIN((T/I-INT(T/I))*2*pi
):
156 PRINT"[HOME]";GOSUB 9070:PRINTF$;
```

```

GOSUB 9070
157 PRINTF$"[2 UP][VERDE]ISTRUZIONI ?";
:GOSUB 9060:IF I$="N" THEN 200
160 PRINT"[CLEAR][BLEU][2 DOWN]INTRODUC
I COSI' I DATI:[2 DOWN][ROSSO]MESE/G
IO/ANNO
175 PRINT"[2 DOWN][ROSSO](8/13/57[BLEU]
=AGO 13,1957)":PRINT"[3 DOWN][VIOLA]
USARE LE BARRE!"
182 GOSUB 9050:PRINT"[CLEAR][DOWN][MIN-
MAI]TT L♠ ♣♦♠\♠, ♠♠ ♠/♥♠ ♠
\♠/
185 PRINT"[ROSSO][2 DOWN]F1[BLEU] /UOVO
QUADRO":PRINT"[DOWN][ROSSO]F7[BLEU]
♠TAMPO ALTRI 20 DI'
186 PRINT"[DOWN][ROSSO]-[BLEU] GIORNI
CRITICI":PRINT"[DOWN][ROSSO]♠[BLEU]
ESCI":PRINT"[DOWN][ROSSO]-[BLEU]
-ONFRONTA":GOSUB 9050
200 PRINT"[MAIUSC.][CLEAR][DOWN][BLEU]S
TAMPA DA":PRINT"(ME/DI/AN)[LEFT][UP
]":GOSUB 1000:IF I1=0 THEN 200
210 M1=I1:D1=I2:Y1=I3:D0=D1:FOR I=1 TO
M1-1:D0=D0+A(I):NEXT
215 D3=Y1*365+D0:FOR I=1 TO Y1-1:D3=D3-
(I/4=INT(I/4)):NEXT
220 PRINT"[HOME][5 DOWN][BLEU]NASCITA
":GOSUB 1000:IF I1=0 THEN 220
225 M=I1:FOR I=1 TO M-1:I2=I2+A(I):NEXT
:GOSUB 800:D=I2:Y=I3
270 PRINT"[DOWN]CONFRONTO?":GOSUB 9060:
R=0:T3=0:IF I$="N" THEN 300
280 R=1:PRINT"[HOME][9 DOWN]NASCITA 2"
:GOSUB 1000:IF I1=0 THEN 280

```

```

290 FOR I=1 TO I1-1:I2=I2+A(I):NEXT:T3=
T:GOSUB 800:GOSUB 820
300 IF T<0 OR T3<0 THEN PRINT"[RVS]SI C
ONTA DALLA TUA":PRINT"[RVS]NASCITA
":GOSUB 1030:GOTO 200
400 PRINT"[CLEAR][2 DOWN][ROSSO]A "MID$(
M$,3*M1-2,3)D1","Y1+1900:PRINT"[DO
WN]HAI VISSUTO PER BEN":PRINT"[LEFT
][DOWN]"T"GIORNI
500 GOSUB 9050:POKE 36879,8:PRINT"[CLEA
R][AZZUR]"LEFT$(F$,9):FOR I=1 TO 20
:PRINT"[NERO]-[AZZUR]":NEXTI
501 PRINTLEFT$(F$,19)"[NERO][RVS]"SP$:T
2=M1
510 PRINTLEFT$(F$,19)"[NERO][RVS]"MID$(
M$,3*M1-2,3)Y1+1900"[RVOFF]":PRINT
LEFT$(F$,20);
600 FOR I=1 TO 3:J(I)=0:R(I)=0:FOR J=0
TO 2:CD(I,J)=0:NEXTJ,I:FOR Q=1 TO 4
0:IF Q/2=INT(Q/2) THEN 618
610 E=D3-1:E=INT(7*(E/7-INT(E/7))+.5)+1
:E$=MID$(STR$(D1),2)
615 PRINT"[BLEU][RVS]"MID$(W$,E,1)"[RVO
FF][DOWN][LEFT]"LEFT$(E$,1)"[DOWN]"
:IF LEN(E$)>1 THEN PRINT"[LEFT]"RI
GHT$(E$,1);
617 PRINT"[2 UP]";
618 POKE 36878,9:FOR I=1 TO 3:X=FNS(18+
I*5):GOSUB 820:R(I)=R(I)-((SGN(X)=
SGN(FNS(18+I*5))) *2.5:GOSUB 820
619 IF X<.6 AND X>-.6 THEN CD(I,J(I))=M
1*100+D1:J(I)=J(I)+1
620 Z=PEEK(FNP(X)):IF Z=32 OR Z=45 OR Z
=P(I)-THEN POKE FNP(X),P(I):POKE FN

```

```

        P(X)+54272,I+3
622 GOTO 630
625 POKE FNP(X),42
630 NEXT:D3=D3+.5:T=T+.5:T3=T3+.5:D1=D1
    +.5
640 IF D1>=A(M1)+1 THEN D1=1:M1=M1+1:IF
    M1>12 THEN M1=1:Y1=Y1+1
650 NEXT:POKE 198,0:IF T2<>M1 AND DD1<>
    1 THEN PRINTLEFT$(F$,19)"[11 RIGHT]
    [NERO][RVS]"MID$(M$,3*M1-2,3)Y1+190
    0
655 PRINTF$"[VERDE]PREMI F1,F7,C,Q)":IF
    R=1 THEN PRINT",R";
730 GOSUB 9060:I=ASC(I$):IF I=133 THEN
    200
735 IF I=81 THEN PRINT"[CLEAR][VERDE]":
    POKE 53281,0:END
745 IF I<>82 OR R=0 THEN 750
746 PRINT"[CLEAR][BIANCO]COMPATIBILITA"
    :PRINT"[DOWN]FIS: "R(1)"%":PRINT"EM
    OT:"R(2)"%":PRINT"INT: "
750 IF I=136 THEN 400
780 IF I<>67 THEN 655
785 PRINT"[CLEAR][ROSSO][RVS]GIORNI CRI
    TICI ":PRINT"[2 DOWN][RVS]P[RVOFF]S
    IC.[2 RIGHT][RVS]E[RVOFF]MOT.[2 RIG
    HT][RVS]I[RVOFF]NT."
790 FOR I=1 TO 3:FOR J=0 TO 2:IF CD(I,J
    )=0 THEN 794
791 K2=INT(CD(I,J)/100):K1=CD(I,J)-K2*1
    00:K2=K2-1
792 PRINTLEFT$(F$,6+2*J) TAB((I-1)*7)MI
    D$(B$,K2*3+1,3)STR$(INT(K1))
794 NEXTJ,I:GOTO 655

```



```

800 T=0:IF Y1=I3 THEN 810
805 FOR I=I3+1 TO Y1-1:T=T-(I/4=INT(I/4
  )):NEXT:T=T-(I3/4=INT(I3/4) AND I2>
  58)
810 T=T+(Y1-I3-1-(D0>=I2))*365-(D0>=I2)
  *(D0-I2)-(D0<I2)*(365-I2+D0):RETURN
820 K2=T:T=T3:T3=K2:RETURN
1000 I=4:INPUT I$:I1=VAL(LEFT$(I$,2)):IF
  MID$(I$,2,1)>"9" OR MID$(I$,2,1)<"
  0" THEN I=3
1005 I2=VAL(MID$(I$,1,2)):I3=VAL(RIGHT$(
  I$,2)):A(2)=FNY(I3):IF I1>12 OR I1<
  1 THEN 1030
1020 IF I2>0 AND I2<=A(I1) THEN RETURN
1030 PRINT"[RVS]NON VALIDO.  ":FOR I=0 T
  O 1500:NEXT:PRINT"[CLEAR]":I1=0:RET
  URN
9050 PRINTF$("[UP][BLEU] PREMI UN TASTO"
9060 GET I$:IF I$="" THEN 9060
9065 RETURN
9070 FOR I=1 TO 21:PRINTMID$("[NERO][BIA
  NCO][ROSSO][AZZUR][VIOLA][VERDE][BL
  EU][GIALLO]",I+(I>8)*8+(I>16)*8,1)"
  [RVS] "):NEXT:PRINT"[VERDE][RVS][LE
  FT][INST] [RVOFF]":RETURN
9100 DATA 31,28,31,30,31,30,31,31,30,31,
  30,31

```

DIFFERENZA DATE

* *

Quanti giorni ci sono da oggi al primo gennaio del 2000?

Chiedetelo al vostro computer e lui saprà dirvelo con esattezza, e.....
velocemente!

```
10 PRINT"[CLEAR][2 DOWN][RVS]  NUMERO  
    DI GIORNI          TRA 2 DATE.  
    (GG,MM,AA)        [RVOFF]"  
20 PRINT  
30 PRINT"PRIMA DATA"  
40 INPUT D1,M1,Y1  
45 PRINT  
50 PRINT"SECONDA DATA"  
60 INPUT D2,M2,Y2  
70 M=M1  
80 D=D1  
90 Y=Y1  
100 GOSUB 230  
110 N=A  
120 M=M2  
130 D=D2  
140 Y=Y2  
150 GOSUB 230  
160 N=A-N  
165 PRINT
```

```

170 PRINT"DIFFERENZA = ":PRINTN" GIORNI
"
180 PRINT
190 PRINT"ALTRI DATI (S/N)":
200 INPUT X$
210 IF X$="S" THEN 20
220 GOTO 460
230 ON MGOTO 260,280,260,340,260,340,26
    0,260,340,260,340,260
240 PRINT"DATA IRREALE"
250 RETURN
260 IF D>31 THEN 240
270 GOTO 350
280 IF Y/4<>INT(Y/4) THEN 310
290 IF Y/400=INT(Y/400) THEN 320
300 IF Y/100<>INT(Y/100) THEN 320
310 IF D>28 THEN 240
320 IF D>29 THEN 240
330 GOTO 350
340 IF D>30 THEN 230
350 DATA 0,31,59,90,120,151,181,212
351 DATA 243,273,304,334
360 RESTORE
361 FOR Q=1 TO M
362 READ A
363 NEXTQ
380 A=A+Y*365+INT(Y/4)+D+1-INT(Y/100)+I
    NT(Y/400)
390 IF INT(Y/4)<>Y/4 THEN 450
410 IF Y/400=INT(Y/400) THEN 430
420 IF Y/100=INT(Y/100) THEN 440
430 IF M>2 THEN 450
440 A=A-1
450 RETURN
460 END

```

GIORNO DELLA SETTIMANA

★ ★

Che giorno della settimana era il 15 ottobre del 1943? Oppure in che giorno della settimana è nato un vostro più caro amico? La risposta alla velocità del vostro fedele computer.

```
10 PRINT"[CLEAR][2 DOWN][RVS]GIORNO DE  
LLA SETTIMANA[RVOFF]"  
20 PRINT  
30 PRINT"(INSERIRE 0,0,0          PER T  
ERMINARE)"  
35 PRINT  
40 PRINT"GIORNO, MESE, ANNO",  
50 INPUT D,M,Y  
60 IF M<0 THEN 100  
70 IF D<0 THEN 100  
80 IF Y<0 THEN 100  
90 GOTO 350  
100 IF M>2 THEN 130  
110 M=M+12  
120 Y=Y-1
```

```

130 N=D+2*M+INT(.6*(M+1))+Y+INT(Y/4)-IN
    T(Y/100)+INT(Y/400)+2
140 N=INT((N/7-INT(N/7))*7+.5)
150 IF N>0 THEN 180
160 PRINT"[DOWN][RVS]DOMENICA[RVOFF]"
170 GOTO 340
180 IF N>1 THEN 210
190 PRINT"[DOWN][RVS]LUNEDI[RVOFF]"
200 GOTO 340
210 IF N>2 THEN 240
220 PRINT"[DOWN][RVS]MERTEDI[RVOFF]"
230 GOTO 340
240 IF N>3 THEN 270
250 PRINT"[DOWN][RVS]MERCOLEDI[RVOFF]"
260 GOTO 340
270 IF N>4 THEN 300
280 PRINT"[DOWN][RVS]GIOVEDI[RVOFF]"
290 GOTO 340
300 IF N>5 THEN 330
310 PRINT"[DOWN][RVS]VENERDI[RVOFF]"
320 GOTO 340
330 PRINT"[DOWN][RVS]SABATO[RVOFF]"
340 PRINT
350 GOTO 40
360 END

```

IDENTIFICATORE ALFABETICO * *

Il programma converte un numero compreso tra 0 e 26 nel corrispondente carattere alfabetico.

Osservate come per il programma di conversione si faccia uso dei comandi READ e DATA.

```
10 PRINT"[CLEAR]"
20 PRINT"INTRODUCI IL NUMERO DI UNA LET
   TERA "
30 PRINT"DELL'ALFABETO           DA 1 A
   26"
40 INPUT N
50 IF N<1 OR N>26 THEN PRINT"[2 UP]":G
   OTO 40
70 FOR L=1 TO N
80 READ A$
90 NEXT
100 PRINT
110 PRINT"LA LETTERA CORRISPONDENTE AL
   NUMERO";N;"   E' : [RVB]";A$
120 GET A$:IF A$="" THEN 120
200 FOR L=1 TO 17:PRINT:NEXT:PRINT"[CLE
   AR]"
```

```
210 RESTORE
220 GOTO 20
300 DATA A,B,C,D,E,F,G,H,I,J,K,L,M,N,O
      ,P,Q,R,S,T,U,V,W,X,Y,Z
```



PERCENTUALI * *

Di solito è più pratico introdurre percentuali espresse in punti per 100 piuttosto che doverle convertire in decimali.

Naturalmente il computer per lavorare ha bisogno dei valori decimali.

Questo programma trasforma appunto dei valori percentuali in decimali.

```
10 PRINT"[CLEAR]":CLR
20 INPUT "PERCENTUALE: ";P
30 D=0.01*P
40 PRINT"DECIMALE";D
50 PRINT:PRINT:PRINT
60 CLR :GOTO 20
```


SISTEMI LINEARI

★ ★

Come tutti i computer anche il vostro VIC-20 ha delle capacità matematiche non indifferenti.

Ecco come sfruttarle nella risoluzione di un sistema lineare di equazioni ad -n- incognite.

```
200 PRINT"[CLEAR][DOWN]          RISOLUZIONE  
E":PRINT"          _____"  
201 PRINT"          DI  SISTEMI":PRINT"  
          _____"  
202 PRINT" DI EQUAZIONI LINEARI":PRINT"  
          _____"  
300 PRINT"[2 DOWN]DI QUANTE EQUAZIONI S  
ICOMPONE IL SISTEMA"  
301 INPUT N  
310 IF N<1 THEN PRINT"[DOWN]IMPOSSIBILE  
! RIPETI!":GOTO 301  
320 IF N>15 THEN PRINT"[DOWN]MI DISPIAC  
E, NON          ACCETTO PIU' DI          0  
UINDICI EQUAZIONI":END  
330 DIM A(N,N),R(N),V(N)  
340 PRINT"[DOWN]E";N;"INCOGNITE"  
350 PRINT"SARANNO INDICATE CON  X1...X"  
      ;MID$(STR$(N),2,10)  
360 GOSUB 900:FOR J=1 TO N
```

```

370 PRINT"SCRIVI I PARAMETRI      DELL'EQ
    UAZIONE";J
380 PRINT:FOR K=1 TO N
390 PRINT"COEFFICIENTE DI X";MID$(STR$(K
    ),2,10)
400 INPUT A(J,K):PRINT:NEXT
410 INPUT "TERMINE NOTO";R(J)
420 GOSUB 900:NEXT
430 GOSUB 2000
500 PRINT"LA SOLUZIONE E'":PRINT
510 FOR J=1 TO N
520 PRINT" X";MID$(STR$(J),2,10);"=";V(
    J)
530 NEXT:END
900 PRINT:PRINT" -----"
910 PRINT:RETURN
2000 IF N=1 THEN V(1)=R(1)/A(1,1):RETURN
2010 FOR K=1 TO N-1
2020 I=K+1
2030 L=K
2040 IF ABS(A(I,K))>ABS(A(L,K)) THEN L=I
2050 IF I<N THEN I=I+1:GOTO 2040
2060 IF L=K THEN 2100
2070 FOR J=K TO N:Q=A(K,J):A(K,J)=A(L,J)
2080 A(L,J)=Q:NEXT
2090 Q=R(K):R(K)=R(L):R(L)=Q
2100 I=K+1
2105 IF A(K,K)=0 THEN PRINT"C'E' UN ERROR
    E;          IL SISTEMA E'          [RVS
    ]IMPOSSIBILE[RVOFF]"
2106 IF A(K,K)=0 THEN END
2110 Q=A(I,K)/A(K,K):A(I,K)=0
2120 FOR J=K+1 TO N:A(I,J)=A(I,J)-Q*A(K,
    J):NEXT

```

```

2130 R(I)=R(I)-Q*R(K):IF I<N THEN I=I+1:
      GOTO 2110
2140 NEXT I
2145 IF A(N,N)=0 THEN PRINT"C'E'UN ERROR
      E' IL SISTEMA E'[RVS]IMPOSSIBILE[RV
      OFF]"
2146 IF A(N,N)=0 THEN END
2150 V(N)=R(N)/A(N,N):FOR I=N-1 TO 1 STE
      P -1
2160 Q=0:FOR J=I+1 TO N:Q=Q+A(I,J)*V(J)
2170 V(I)=(R(I)-Q)/A(I,I):NEXT J:NEXT I
2180 RETURN

```

ORDINAMENTO ALFABETICO * *

All'inizio vi verrà domandato quanti elementi numerici o alfabetici volete mettere in ordine alfabetico.

In seguito la visualizzazione avverrà in gruppi da 20 elementi.

Premete il tasto (c) per continuare la visualizzazione.

```
10 PRINT"[CLEAR][2 DOWN][RVS]ORDINAMEN  
   TO ALFABETICO[RVOFF]"  
20 PRINT  
30 PRINT"--INTRODURRE 0 PER          TERMIN  
   ARE IL          PROGRAMMA"  
35 PRINT  
40 PRINT"NUMERO DI VOCI":  
50 INPUT N  
60 IF N=0 THEN 330  
70 DIM A$(N)  
80 FOR I=1 TO N  
90 PRINT"VOCE";I;  
100 INPUT A$(I)  
110 NEXT I  
115 PRINT"[CLEAR][RIGHT]"  
120 M=N  
130 T=M/2
```

```

131 M=INT(T)
140 IF M=0 THEN 280
150 K=N-M
160 J=1
170 I=J
180 L=I+M
190 IF A$(I)<=A$(L) THEN 250
200 T$=A$(I)
210 A$(I)=A$(L)
220 A$(L)=T$
230 I=I-M
240 IF I>=1 THEN 180
250 J=J+1
260 IF J>K THEN 130
270 GOTO 170
280 FOR I=1 TO N
282 IF I/20<>INT(I/20) THEN 290
284 PRINT"[RVS] (C) PER CONTINUARE [RVO
FF]"
285 GET C$:IF C$<>"C" THEN 285
290 PRINTA$(I)
300 NEXTI
310 PRINT
320 CLR :GOTO 40
330 END

```

DA ESADECIMALE IN DECIMALE * *

I primi calcolatori inventati dall'uomo "masticavano" solo numeri binari. Noi siamo abituati a contare in base dieci.

Oggi la tecnologia del silicio ha generalmente adottato la base 16 per velocizzare i calcoli meccanici.

Questo programma vi permetterà di acquistare familiarità con i numeri esadecimali, trasformandoli in numeri decimali, per noi di più immediata comprensione.

```
10 F2=4096:F3=256:F4=16
20 PRINT"[CLEAR]CONVERSIONE
   DA ESADECIMALE      IN DECIMALE
   MAX: (.....)"
30 INPUT Y$:IF LEN(Y$)>4 THEN RUN
35 IF LEN(Y$)<4 THEN PRINT"[UP]AGGIUNG
   I GLI ZERI PER LE CIFRE NON
   SIGNIFICATIVE":GOTO 2000
40 A$=LEFT$(Y$,1):X$=A$:GOSUB 1000
50 A=S*F2
60 B$=MID$(Y$,2,1):X$=B$:GOSUB 1000
70 B=S*F3
80 C$=MID$(Y$,3,1):X$=C$:GOSUB 1000
```

```

90 C=S*F4
100 D$=RIGHT$(Y$,1):X$=D$:GOSUB 1000
110 D=S
120 T=A+B+C+D:PRINT"[UP] ",T
130 GET Z$:IF Z$="" THEN 130
150 GOTO 30
1000 IF X$="A" THEN S=10:RETURN
1010 IF X$="B" THEN S=11:RETURN
1020 IF X$="C" THEN S=12:RETURN
1030 IF X$="D" THEN S=13:RETURN
1040 IF X$="E" THEN S=14:RETURN
1050 IF X$="F" THEN S=15:RETURN
1060 S=VAL(X$)
1070 RETURN
2000 FOR GG=1 TO 2000:NEXT: RUN

```

NUMERI DEI CARATTERI

★ ★

Questo breve programma mostra il valore ASCII per ogni carattere della tastiera, accanto al carattere che rappresenta.

Sarai rapidamente in grado di dire quale carattere corrisponda ogni numero del codice ASCII.

Nella linea 40 c'è un temporizzatore per permettere una lettura a tempo delle informazioni.

Agendo sul numero 400 della linea 40 è possibile abbreviare o allungare il tempo di permanenza delle immagini sullo schermo.

```
10 PRINT"[CLEAR]"
20 FOR N=33 TO 127
30 PRINTN,CHR$(N):PRINT
40 FOR T=1 TO 400:NEXT T
50 IF N=255 THEN 20
60 NEXT N
70 FOR N=161 TO 255
80 GOTO 30
```


SIMULATORE DI CALCOLATRICE RPN

★ ★


Il vostro computer può essere predisposto per lavorare come una calcolatrice funzionante col sistema RPN (il termine RPN non è altro che l'acronimo di Notazione Polacca Inversa).

Per eseguire un calcolo digitate un numero seguito immediatamente dal simbolo di un operatore aritmetico.

Il tasto STOP pone fine al programma.

```
10 PRINT"[CLEAR]":REM  PULIZIA SCHERMO
20 N$="":X$=""
25 PRINTT$
35 GET N$:IF N$="" THEN 35
40 IF ASC(N$)<42 OR ASC(N$)>57 THEN 35
50 IF ASC(N$)=46 OR ASC(N$)>47 THEN X$
   =X$+N$:GOTO 35
60 IF ASC(N$)=42 THEN 1100
65 IF ASC(N$)=43 THEN 1200
70 IF ASC(N$)=45 THEN 1300
80 IF ASC(N$)=47 THEN 1400
90 GOTO 10
1100 X=VAL(X$):T=T*X
1120 GOTO 1420
```

```
1200 X=VAL(X$):T=T+X  
1220 GOTO 1420  
1300 X=VAL(X$):T=T-X  
1320 GOTO 1420  
1400 X=VAL(X$):T=T/X  
1420 T$=STR$(T)  
1430 GOTO 10
```



INVERTITORE DI NUMERI

★ ★

Dai al tuo computer un qualsiasi numero di tre cifre e come risultato lo rivedrete capovolto.

Per esempio 234 vi verrà trasformato in 432.

```
10 PRINT" [CLEAR] ":CLR
20 PRINT"INTODUCI UN NUMERO DI 3 CIFRE"
"
30 INPUT N
40 IF N<100 OR N>999 THEN PRINT"[2 UP]"
" :GOTO 30
50 A=INT(N/100)
60 B=INT(10*((N/100)-A))
70 C=VAL(RIGHT$(STR$(N),1))
80 PRINT:PRINTRIGHT$(STR$(C),1);RIGHT$(
  (STR$(B),1);RIGHT$(STR$(A),1)
90 PRINT:PRINT:PRINT
100 CLR :GOTO 20
```

INTEGRALI * * *

Questo utile programma permette il calcolo di integrali anche molto complessi.

Di volta in volta dovreste inserire nella linea specificata dal programma la funzione di cui vorrete studiare gli intervalli di integrazione.

```
8 POKE 36879,8:PRINT"[VERDE2]"
150 N=2
160 V=221
200 PRINT"[CLEAR] INTEGRALE CALCOLATO
      CON LA FORMULA":PRINTSPC(4)" DI
      SIMPSON"
220 PRINT"ATTENZIONE"
240 PRINT"LA SUBROUTINE          ALLA LI
      NEA 5000-5999  DEFINISCE"
260 PRINT"Y COME FUNZIONE DI X"
300 PRINT:PRINT"LIMITESUPERIORE DI X":
      INPUT L
310 PRINT:PRINT"LIMITESUPERIORE DI X":
      INPUT U
360 PRINT"N.SEZ.", "INTEG."
400 DX=(U-L)/N:T=0
410 X=L:GOSUB 5000:T=T+Y
420 X=U:GOSUB 5000:T=T+Y
450 M=N/2:Z=0
```

```

460 FOR J=1 TO M
470 X=L+DX*(2*J-1):GOSUB 5000:Z=Z+Y
480 NEXT:T=T+4*Z
500 M=M-1:IF M=0 THEN 600
510 Z=0:FOR J=1 TO M
520 X=L+DX*2*J:GOSUB 5000:Z=Z+Y
530 NEXT:T=T+2*Z
600 A=DX*T/3
610 PRINTN,A
620 N=N*2
630 GOTO 400
4970 REM ** ISERISCI LA SUBROUTINE A 500
    0
4980 REM
4990 REM
5000 REM  Y=F(X) VA QUI'
5999 RETURN

```

DEVIAZIONE STANDARD

★ ★

Ecco come determinare qualsiasi deviazione standard relativa ad una certa media.

Per uscire dal programma occorre inserire il valore 999999999.

```
10 PRINT"ICLEARI":CLR
20 INPUT "VALORE: ";X
30 IF X=999999999 THEN 60
40 T=T+X:S=S+X^2:N=N+1
50 GOTO 20
60 A=T/N:V=S/N-A^2:D=SQR(V)
70 PRINT:PRINT:PRINT
80 PRINT"TOT. VALORI INSERITI: ";T
90 PRINT"MEDIA: ";A
100 PRINT"VARIANZA: ";V
110 PRINT"DEVIAZ. STANDARD: ";D
120 PRINT:PRINT:PRINT
130 PRINT"PREMI UN TASTO PER CONTINUARE"
140 GET H$:IF H$="" THEN 140
150 CLR :GOTO 10
```

DIVISIONE DI FATTORI

★ ★

Questo programma trova tutti i fattori primi in cui risulta divisibile un numero qualunque.

Può essere usato come sottoprocedura di un programma più lungo.

Il numero di fattori individuali è limitato dalla istruzione DIM.

```
10 PRINT"[CLEAR]": REM  PULIZIA SCHERM
   0
20 DIM Q(650)
30 INPUT "NUMERO MAX.1250":N
40 FOR L=2 TO N/2
50 M=N/L
60 IF M=INT(M) THEN Q(L)=M
70 NEXTL
80 PRINT:PRINT"I FATTORI SONO:"
90 FOR L=1 TO N/2
100 IF Q(L)>1 THEN PRINTQ(L):GOTO 120
110 Z=Z+1
120 NEXTL
130 IF N=1 THEN PRINT"NESSUNO":GOTO 150
140 IF Z=INT(N/2) THEN PRINT"NESSUNO"
150 GET A$:IF A$="" THEN 150
160 RUN
```

FUNZIONI LOGICHE

★ ★

Puoi far fare al computer cose basate sulla sua decisione che qualcosa possa esistere.

Questo primo programma stampa il valore di C solo se B esiste.

La decisione è nella linea 40.

La macchina stampa B solo se non è eguale a 0.

A partire dalla linea 100 si ha un secondo programma il quale visualizza i risultati delle prime linee solo se risulta "vero" uno o entrambi i confronti.

```
0 REM PRIMO ESEMPIO
10 PRINT"[CLEAR]":CLR
20 B=10
30 C=10*B
40 IF B THEN PRINTC
99 END
100 REM SECONDO ESEMPIO
110 PRINT"[CLEAR]":CLR
120 B=(6+8)>(3*4)
130 C=(5+2)>(9+2)
140 IF B THEN PRINT"B OKAY"
150 IF C THEN PRINT"C OKAY"
200 END
```


STATISTICHE IN BASE 1000

★ ★

Supponete che la vostra radio locale affermi di raccogliere 51000 ascoltatori per ogni 133000 lire di spese pubblicitarie e che l'amministratore di un quotidiano sostenga che il costo di un'inserzione sia di 330000 lire per ogni 1000 lettori.

Il seguente programma fornisce i valori necessari per stabilire quale delle due pubblicità sia più redditizia, valutando la percentuale in base mille.

```
10 PRINT "[CLEAR]"
20 PRINT "COSTO INSERZIONE": INPUT A
30 PRINT "CIRCOLAZIONE": INPUT C
40 M=1000*(A/C)
45 PRINT
50 PRINT "£"/M
55 PRINT "COSTO PER MILLE"
57 GET A$: IF A$="" THEN 57
60 FOR Q=1 TO 20:PRINT:NEXT Q
70 GOTO 10
```

NUMERI CASUALI: ORDINAMENTO MAGGIORE/MINORE

★ ★

E' importante poter ordinare un gruppo di numeri e stabilire quali siano i valori maggiori e quelli minori.

Questa funzione è svolta dal seguente programma.

Il generatore di numeri casuali si trova alla linea 30.

Il range va da 0 a 999.


La linea 50 determina il numero più basso, mentre la linea 60 il più alto della serie.

```
10 PRINT"ICLEARJ":X=RND(-TI)
20 FOR L=0 TO 99
30 N=INT(1000*RND(1))
40 IF L=0 THEN LN=N:HN=N
50 IF N<LN THEN LN=N
60 IF N>HN THEN HN=N
70 PRINT"STO SCEGLIENDO"
80 NEXTL
90 PRINT"ICLEARJ": REM PULIZIA SCHERM
  0
100 PRINT"NUMERO BASSO:",LN
```

```

110 PRINT"NUMERO ALTO:",HN
120 PRINT:PRINT:PRINT:PRINT
130 PRINT"PER AVERNE UN ALTRO"
140 PRINT"PREMI UN TASTO          QUALSIA
    SI"
150 GET K$
160 IF K$="" THEN 150
170 CLR
180 GOTO 10

```



MEDIA NUMERICA CASUALE * *

Lo scopo di questo programma è calcolare la somma e la media di 100 numeri generati casualmente tra 0 e 9.

Per le persone più interessate all'argomento, il programma può rappresentare lo spunto teorico per un approfondimento delle teorie probabilistiche.

```
10 PRINT"[CLEAR]":X=RND(-TI)
20 FOR L=0 TO 99
30 N=INT(10*RND(1))
40 NT=NT+N
50 PRINT"STO FACENDO LA MEDIA"
60 NEXTL
70 PRINT"[CLEAR]":REM PULIZIA SCHERM
  0
80 AV=NT/100
90 PRINT"IL TOTALE DI 100"
100 PRINT"NUMERI A CASO"
110 PRINT"DA 0 A 9"
120 PRINT"E"/;NT
130 PRINT
140 PRINT"LA MEDIA E"/;AV
150 PRINT:PRINT:PRINT:PRINT
```

```
160 INPUT "PER PROSEGUIRE PREMI  RETURN  
    ";K$  
170 CLR :GOTO 10
```



NUMERO MEDIO DI UNA SERIE * *

Ecco come trovare il numero centrale di una serie di numeri.
Inserite una serie di numeri e uno zero dopo l'ultimo numero.
Le linee 40 e 70 selezionano il numero più alto e il più basso della serie.
La linea 90 il valore medio della serie.

```
10 PRINT"[CLEAR]": REM  PULIZIA SCHERM
   0
20 PRINT"DAMMI UN NUMERO":INPUT Z
30 IF Z=0 THEN 90
40 N=N+1
50 IF N=1 THEN H=Z:L=Z
60 IF Z<L THEN L=Z
70 IF Z>H THEN H=Z
80 GOTO 20
90 M=L+((H-L)/2)
100 PRINT:PRINT:PRINT
110 PRINT"NUMERO A META' DELLA SERIE:"
    M
120 PRINT:PRINT:PRINT
130 INPUT "PER PROSEGUIRE PREMI RETURN
    " :K$
140 CLR :GOTO 10
```

NUMERI CASUALI: DISTRIBUZIONE

★ ★

Vi siete mai domandati quanto siano effettivamente casuali i numeri prodotti da un generatore RND?

Il seguente programma genera 100 numeri casuali tra 0 e 9 e conta quante volte viene ripetuto ogni valore.

Alla fine viene mostrato un quadro riassuntivo dei risultati.

```
10 PRINT"DCLEARJ":X=RND(-TI)
20 FOR L=1 TO 100
30 N=INT(10*RND(1))
40 IF N=0 THEN A=A+1
50 IF N=1 THEN B=B+1
60 IF N=2 THEN C=C+1
70 IF N=3 THEN D=D+1
80 IF N=4 THEN E=E+1
90 IF N=5 THEN F=F+1
100 IF N=6 THEN G=G+1
110 IF N=7 THEN H=H+1
120 IF N=8 THEN I=I+1
130 IF N=9 THEN J=J+1
140 PRINT"STO CALCOLANDO"
150 NEXTL
160 PRINT"DCLEARJ": REM PULIZIA SCHERM
```

```
0
170 PRINT"0",A
180 PRINT"1",B
190 PRINT"2",C
200 PRINT"3",D
210 PRINT"4",E
220 PRINT"5",F
230 PRINT"6",G
240 PRINT"7",H
250 PRINT"8",I
260 PRINT"9",J
270 PRINT:PRINT:PRINT
280 INPUT "PER PROSEGUIRE PREMI. RETURN
";K$
290 CLR :GOTO 10
```


CLASSIFICAZIONE

★ ★

Un gran quantitativo di numeri può essere classificato e diviso in una quantità più piccola.

Guarda il nostro esempio:

accoglie risultati e li divide in categorie denominate A,B,C,D ed F.

Il programma presume che i singoli valori varino in un range compreso fra 0 e 100.

La classificazione in lettere fa corrispondere una F ai risultati da 0 a 59

D ai valori 60-69

C ai valori 70-79

B ai valori 80-89 e A ai punteggi dal 90 al 100.

Immetti tutti i punteggi che vuoi e infine la lettera X per concludere il ciclo di immissione.

Le linee 100-140 dividono tutti i punteggi nelle categorie A-F.

Le linee 150-170 segnalano i valori massimi e minimi mentre la 200 trova il valore medio e quello più probabile.

```
10 PRINT"[CLEAR]"
20 PRINT"DIGITA UN GRUPPO DI DATI"
30 PRINT"DA 1 A 100 , UNO ALLA VOLTA"
40 PRINT:PRINT:PRINT
50 PRINT"INSERISCI X PER FINIRE"
60 PRINT:PRINT
80 INPUT "[RVS]DATO[RVOFF]";G$
```

```

85 IF G$="" THEN 80
90 IF G$="X" THEN 200
100 G=VAL(G$)
110 N=N+1
120 IF G<60 THEN F=F+1:GOTO 150
125 IF G<70 THEN D=D+1:GOTO 150
130 IF G<80 THEN C=C+1:GOTO 150
135 IF G<90 THEN B=B+1:GOTO 150
140 A=A+1
150 IF N=1 THEN L=G:H=G
160 IF G<L THEN L=G
170 IF G>H THEN H=G
180 S=S+G
185 G$=""
190 GOTO 80
200 P=S/N:M=L+((H-L)/2)
210 PRINT"[CLEAR]"
220 PRINT"CI SONO ";N;"DATI"
230 PRINT"I VALORI VANNO DA";L:PRINT" A
    ";H
240 PRINT"IL VALORE CENTRALE E'";M
250 PRINT"LA MEDIA E'";P
260 PRINT
270 PRINT"TOTALE PER OGNI          CATEGOR
    IA"
280 PRINT
290 PRINT"A",A
300 PRINT"B",B
310 PRINT"C",C
320 PRINT"D",D
330 PRINT"F",F
410 PRINT"PREMI [RVS]RETURN[RVOFF] PER
    CONTINUARE"
420 INPUT K$

```

```
430 CLR  
440 GOTO 10
```



DECIMALE BINARIO

★ ★

I computer della nostra generazione sono basati su una logica a 2 stati (aperto—chiuso). E' importante quindi acquisire una approfondita conoscenza sui sistemi di numerazione differenti dal nostro (decimale). Il programma che vi presentiamo consente di effettuare conversioni dal sistema decimale a quello binario.

```
100 REM DEC-BIN
150 INPUT "[CLEAR]NUMERO DECIMALE";A
160 IF A>255 THEN PRINT"[RVS]1ERRORE[RVS]
    OFF]":FOR P=0 TO 1000:NEXT:GOTO 150
195 PRINT"TRASFORMAZIONE IN ATTO"
200 FOR I=7 TO 0 STEP -1:IF A>=2↑I THEN
    B$=B$+"1":A=A-2↑I:GOTO 300
250 B$=B$+"0"
300 NEXT
500 PRINT TAB(7);B$
510 PRINT"[DOWN]ANCORA ? (S/N)"
520 GET A$:IF A$="" THEN 520
530 IF A$="S" THEN RUN
540 END
```

BINARIO DECIMALE

★ ★

Questo programma rappresenta un complemento del precedente, in quanto consente conversioni dal sistema binario a quello decimale.

```
100 REM: BIN-DEC
150 PRINT"[CLEAR]NUMERO BINARIO"
155 INPUT ">>>>":A$
160 IF LEN(A$)<>8 THEN PRINT"[RVS]1ERRO
RE[RVOFF]":FOR P=0 TO 1000:NEXT:GOT
O 150
195 PRINT"[DOWN]TRASFORMAZIONE IN ATTO"
200 TL=0:C=0
210 FOR X=8 TO 1 STEP -1:C=C+1
220 Y=VAL(MID$(A$,C,1))
230 TL=TL+Y*2^(X-1)
240 NEXTX
500 PRINT TAB(10);TL
510 PRINT"[DOWN]ANCORA ? (S/N)"
520 GET A$:IF A$="" THEN 520
530 IF A$="S" THEN RUN
540 END
```

EQUIVALENZE * *

Ecco un quiz che metterà a dura prova le vostre conoscenze sui sistemi di misura. Il numero delle domande a cui si vuole essere sottoposti è prefissabile dall'utente.

```
150 DIM ES$(35),MS$(35),C(35),EP$(35),M
    P$(35)
160 Q=RND(-TI):B$=" "
200 GOSUB 400:GOSUB 450
210 INPUT " QUANTE DOMANDE VUOI";NQ:NQ=
    INT(NQ):IF NQ<1 THEN 210
220 FOR J=1 TO NQ:GOSUB 600:GOSUB 900:N
    EXT
230 GOSUB 450:PRINT"[ROSSO]HAI DATO";NR
    ;"RISPOSTE GIUSTE SU";NQ;"DOMANDE":
    PRINT:PRINT"[BLEU]"
240 P=100*NR/NQ:PRINT"PERCENTUALE RISPO
    STE GIUSTE =";P
250 END
400 RESTORE :ND=0
410 ND=ND+1:READ ES$(ND),MS$(ND),C(ND),
    EP$(ND),MP$(ND)
420 IF ES$(ND)<>"XXX" THEN 410
430 ND=ND-1:RETURN
450 PRINTCHR$(147); TAB(3);"[IRVSIUN QUI
```

```

      Z METRICO":PRINT:PRINT"*****
      *****":RETURN
600 N=INT(RND*10)+1
610 F=0:IF RND(1)>0.5 THEN F=1
620 V1=INT(RND(1)*99)+2:V3=V1*C(N):IF F
    =1 THEN V3=V1/C(N)
630 IF N1 THEN V3=(V1-32)/1.8:IF F=1 TH
    EN V3=(V1*1.8)+32
640 V2=V3*(0.5+RND(1)):V2=INT(V2+0.5):T
    =0:IF V2<V3 THEN T=1
650 GOSUB 450:PRINT"DOMANDA";J;"SU":N0:
    PRINT
660 IF F=0 THEN PRINTV1;EP$(N);"
    SONO PIU' DI";V2;MP$(N);"?"
670 IF F=1 THEN PRINTV1;MP$(N);"
    SONO PIU' DI";V2;EP$(N);"?"
680 GET Q$:IF Q$="" THEN 680
690 IF Q$="S" THEN PRINT:PRINT"[RVS]TU
    DICI SI:[RVOFF]";R=1:GOTO 730
710 IF Q$="N" THEN PRINT:PRINT"[RVS]TU
    DICI NO: [RVOFF]";R=0:GO
    TO 730
720 PRINT:PRINT"[ROSSO] NON LO SAI ":PR
    INT"[BLEU]";R=2
730 X=T-R:IF R=2 THEN GOSUB 800:GOTO 76
    0
740 IF X=0 THEN PRINT:PRINT"[RVS] ED
    HAI RAGIONE! [RVOFF]";NR=NR+1:GOTO
    760
750 PRINT"[RVS][ROSSO] SEI IN ERRORE [N
    ERO]STUDIA[RVOFF]":PRINT"[BLEU]":GO
    SUB 800
760 RETURN
800 PRINT:PRINT"-----"

```

```

      :PRINT
810 IF F=0 THEN PRINTV1;EP$(N);" EQUIVA
    LGONO A ":PRINTV3;MP$(N)
820 IF F=1 THEN PRINTV1;MP$(N);" EQUIVA
    LGONO A ":PRINTV3;EP$(N)
830 PRINT:PRINT"**** LA REGOLA E' ****"
    :PRINT
840 IF N=1 AND F=0 THEN PRINT" GRADI C.
    =(GRADI F.-32/1.8":RETURN
850 IF N=1 AND F=1 THEN PRINT" GRADI F.
    =(GRADI C.*1.8)+32":RETURN
860 IF F=0 THEN PRINT" 1 ";ES$(N);" EQU
    IVALE A":PRINTC(N);MP$(N):RETURN
870 Q=INT(1.E5/C(N))/1.E5:PRINT" 1 ";MS
    $(N);"EQUIVALE":PRINTQ;EP$(N):RETUR
    N
900 PRINT:PRINT:PRINT"[RVS][ROSSO]  BA
    TTI UN TASTO [RVOFF][BLEU]"
910 GET Q$:IF Q$="" THEN 910
920 RETURN
1000 DATA GRADO FAHRENHEIT,GRADO CENTIG
    RADO,0.5
1010 DATA GRADI FAHRENHEIT,GRADI CENTIG
    RADI
1020 DATA MIGLIO ALL'ORA,CHILOMETRO ALL
    'ORA,1.60935
1030 DATA MIGLIA ALL'ORA,CHILOMETRI ALL
    'ORA
1040 DATA PIEDE,METRO,0.3048
1050 DATA PIEDI,METRI
1060 DATA MIGLIO,CHILOMETRO,1.60935
1070 DATA MIGLIA,CHILOMETRI
1080 DATA POLLICE,CENTIMETRO,2.54
1090 DATA POLLICI,CENTIMETRI

```


1100 DATA GALLONE, LITRO, 3.78533
1110 DATA GALLONI, LITRI
1120 DATA LIBBRA, CHILOGRAMMO, 0.45359
1130 DATA LIBBRE, CHILOGRAMMI
1999 DATA XXX, XXX, 0, XXX, XXX



SEQUENZE * *

Provate ora a testare le vostre attitudini al ragionamento.

Riuscirete ad eccellere, se avete doti logiche particolarmente sviluppate.

Il gioco consiste nel trovare l'elemento mancante di una sequenza di numeri.

```
2 PRINT"[CLEAR][DOWN][RVS]  **  SEQU
  ENZE  **  "
3 PRINT"COMPLETATE LA SEQUENZADI NUME
  RI CHE STO PER MOSTRARVI[BLEU][DOWN
  ]"
4 FOR T=1 TO 1000:NEXT
10 LET B=INT(RND(1)*10)+1
20 LET C=INT(RND(1)*10)+1
30 LET D=INT(RND(1)*10)+1
100 FOR A=0 TO INT(RND(1)*6)+4
110 PRINTB
115 B=B+(B+C)-D*B
120 NEXT
125 PRINT" ....."
130 INPUT "[DOWN]NUMERO MANCANTE";X
140 IF X=B THEN PRINT"[DOWN]ESATTO, ERA
  ";B:GOTO 200
150 PRINT"[DOWN]ERRORE, ERA";B
200 INPUT "[DOWN]UN'ALTRA S/N";A$
```

```
210 IF A$="S" THEN 2
220 PRINT"[CLEAR][DOWN][AZZUR][RVS]  ♥♥
      SEQUENZE  ♥♥  "
230 PRINT"[BLEU][4 DOWN]ARRIVEDERCI!!":
      END
```



TROVA IL NUMERO

★ ★

Eccolo!

Da qui, negli anni settanta, cominciò la programmazione dei microcomputer.

Il primo gioco ad essere giocato fu una routine di ricerca numerica con informazioni troppo grande/troppo piccolo.

Il computer seleziona un numero segreto.

Tu devi indovinarlo.

Lo schermo ti informa se il numero che stai provando è troppo elevato, troppo basso o se è proprio quello giusto.

Ecco come funziona:

la linea 100 genera un numero casuale tra zero e 1000 (il numero segreto) e lo immagazzina.

La linea 210 ti chiede di indovinarlo.

Le linee 300-310 decidono se l'hai trovato o meno, ed infine la linea 220 tiene conto dei numeri dei tentativi.

```
10 PRINT"[CLEAR]":REM  PULISCE SCHERMO
20 Z=RND(-TI)
30 T=0
100 R=INT(1000*RND(1))
200 PRINT"HO UN NUMERO SEGRETO ETU DEVI
    SCOPRILO"
210 PRINT"INSERISCI UN NO.":INPUT B
```

```
220 T=T+1
230 PRINT:PRINT:PRINT"TENTATIVO NO. ";T
300 IF B>R THEN PRINT:PRINT"TROPPA ALTO
    , RITENTA":INPUT B:GOTO 220
310 IF B<R THEN PRINT:PRINT"TROPPA BASS
    O, RITENTA":INPUT B:GOTO 220
400 PRINT"[CLEAR]"
410 PRINT"SI, HAI INDOVINATO!"
420 PRINTR;"E' IL NUMERO"
430 PRINT"L'HAI SCOPERTO"
440 PRINT"IN ";T;" TENTATIVI":FOR I=0 TO 5000:NEXT
450 PRINT"[CLEAR]"
460 CLR
470 GOTO 30
```

SCARTA L'INTERO

★ ★

Si consideri il numero 123,456 e si voglia lavorare solo sulla porzione a destra della virgola.

Il programma seguente conserverà solo la parte .456, scartando 123.

In pratica tronca dei numeri decimali, conservandone solo la porzione frazionaria.

```
10 PRINT"[CLEAR]": REM  PULIZIA SCHERM
   0
20 PRINT"DAMMI UN NUMERO"
30 PRINT"CON UNA CIFRA DECIMALE":INPUT
   N
40 X=N-INT(N)
50 PRINT
60 PRINT"LA PORZIONE"
70 PRINT"FRAZIONALE DI"
80 PRINTN;"E/":X
90 PRINT:PRINT:PRINT
100 INPUT "PER PROSEGUIRE PREMI   RETURN
    "OK$
110 CLR :GOTO 10
```

ARROTONDAMENTI

★ ★

La tecnica per arrotondare un numero è semplice.

Questo programma che può operare isolatamente o nel corpo di un programma più vasto, arrotonda un numero decimale all'intero più vicino.

Ci sono due vie per arrotondare un numero.

Una dice:

"Se il numero decimale è maggiore di 5, arrotonda all'intero superiore".

Il che significa che 0.5 arrotonda per difetto.

La seconda è:


"Per ogni numero minore di 5, arrotonda per difetto".


In questo caso 0.5 arrotonda per eccesso.

La prima parte del programma segue la prima teoria, mentre la seconda arrotonda per difetto a meno di 5.

```
10 PRINT"[CLEAR]": REM  PULIZIA SCHERM
   0
20 PRINT"DAMMI UN NUMERO"
30 PRINT"CON PIU' DI"
40 PRINT"DUE CIFRE DECIMALI":INPUT N
50 N$=STR$(N)
60 N$=RIGHT$(N$, (LEN(N$)-1))
70 R=INT(100*N+0.5)/100
80 R$=STR$(R)
```

```
90 R$=RIGHT$(R$, (LEN(R$)-1))
100 PRINT "[CLEAR]"
110 PRINTN:PRINT "SI APPROSSIMA A"
120 PRINTR
130 PRINT:PRINT "      0"
140 PRINT:PRINT "$";N$;" DIVENTA"
150 PRINT "$";R$
160 PRINT:PRINT:PRINT
170 PRINT "PREMI RETURN PER PROSEGUIRE"
175 GET A$:IF A$="" THEN 175
180 CLR :GOTO 10
```





ARROTONDAMENTI A DOPPIA CIFRA


★ ★

E' possibile arrotondare al centesimo più vicino, cioè due cifre dopo la virgola.

Ecco come:

```
10 PRINT"[CLEAR]": REM  PULIZIA SCHERM
   0
20 PRINT"DAMMI UN NUMERO"
30 PRINT"DA APPROSSIMARE          ALL'INT
   ERO"
40 INPUT N
50 IF N>INT(N) THEN 80
60 R=N
70 GOTO 130
80 D=N-INT(N)
90 IF D>0.5 THEN 120
100 R=INT(N)
110 GOTO 130
120 R=INT(N)+1
130 PRINT"[CLEAR]"
140 PRINTN
150 PRINT"E' APPROSSIMATO A"
160 PRINTR
```

170 PRINT:PRINT:PRINT
180 CLR :GOTO 20



DALLA PERCENTUALE AL DECIMALE * *

depositi, gli interessi, le imposte e gli altri programmi finanziari diventano più pratici se non dovete fare conversioni numeriche a mente.

Per esempio, se sapete che il vostro conto bancario vi frutta un'interesse pari all'8%, la via più breve per conoscere l'ammontare finale è di inserire un 8 e lasciare al computer il compito di eseguire i conti.

Inserite nel programma seguente, ad esempio, il valore 2.500.000 e digitate 6 alla richiesta della percentuale.

Il computer risponderà 2.650.000.

```
10 PRINT"[CLEAR]": REM  PULIZIA SCHERM
   0
20 PRINT"PREZZO £":INPUT P
30 PRINT"TASSA DI VENDITA %":INPUT R
40 T=0.01*R
50 S=T*P:B=P+S
60 S$=STR$(S):B$=STR$(B)
70 S$=RIGHT$(S$, (LEN(S$)-1)):B$=RIGHT$(
   B$, (LEN(B$)-1))
80 PRINT"[CLEAR]"
90 PRINT"TASSA DI VENDITA":PRINT"£")S$
100 PRINT"CONTO TOTALE":PRINT"£")B$
110 PRINT:PRINT:PRINT
```

```
120 INPUT "PER PROSEGUIRE PREMI RETURN  
" ; K$  
130 CLR : GOTO 10
```

```
sdfn349zx `1f758g91o  
fjser1zs~dferivera45  
b%73412$15&&5sd?}(s  
gdu1queurwZHkhfkUuxyo  
ohohsdVAL$ jfodfLkfa  
y9wr9sfhop240vcnsjUu  
d.jvnwc43818fhqcnaihv  
s fnxje}si vj 8fhvule  
fhv&vus@v#vh sdjYgecJ  
JkdkheIoejvDAchai198  
s gjz^s8cjN?dbsvwuc*  
*zxc63Fhd zc!fv rZVeg  
vZEx#/vaevetjyuqDhhI  
ibjTwv(gvc<<&?cgrnv!  
\{vbv?xv': ;k959vhi10  
dvhhGIYRrguadighfj fj  
JJrjnIiwpcs?K?i\ aoc>  
=qdic<tqc<=ziJKjfQdc  
Pprf(db&7%ga$3vi#fia  
0c wr8c!cDofjvkweoKk  
fjKQDSdudrgjaNOT av8
```

COMPARATORE DI INTROITI * *

Questo semplice programma raccoglie i dati relativi agli introiti mensili della vostra attività e li visualizza su un grafico di immediata interpretazione, che pone in evidenza come differisca un mese dall'altro.

Le linee dalla 100 alla 160 acquisiscono i dati

le linee dalla 300 alla 360 creano il grafico.

```
10 PRINT"[CLEAR]":REM  PULIZIA SCHERMO
20 DIM CR(6)
100 FOR L=1 TO 6
110 PRINT"ENTRATE DI CASSA"
120 PRINT"DEL MESE":L
130 INPUT CR(L)
140 IF CR(L)>12 THEN PRINT"[2 UP]":GOTO
    130
150 PRINT
160 NEXT
200 PRINT"[CLEAR]"
210 PRINT"[NERO]"
220 PRINT"ENTRATE DI CASSA"
230 PRINT"(VALORI DA 0 A 12)"
240 PRINT"[ROSSO]"
250 PRINT
```

```

300 FOR T=1 TO 6
305 PRINTT;
310 IF CR(T)>12 THEN PRINT"VALORE ECCES
      SIVO"
320 FOR X=1 TO CR(T)
330 IF CR(T)>0 THEN PRINT"LRVSJ ";
340 NEXTX
350 PRINT"LRVOFFJ[DOWN]"
360 NEXTT
400 FOR W=1 TO 3:PRINT:NEXT
410 REM PRINT" "
420 INPUT "PREMI RETURN PER CONTINUARE"
      ;Y$
430 CLR :GOTO 10

```


PUBBLICITA': COSTO PER UNITA' DI VENDITA

★ ★

Il vostro quotidiano preferito possiede il miglior rapporto prezzo quantità per il settore pubblicitario.

Un annuncio vi costa 330000 lire e vi frutta la vendita di 77 prodotti.

Quanto è stato il costo unitario?

La giusta risposta verra data da questo breve programma.

```
10 PRINT"[CLEAR]"
20 PRINT"COSTO INSERZIONE":INPUT A
30 PRINT"UNITA' VENDUTE":INPUT U
40 C=A/U
45 PRINT
50 PRINT"LA PUBBLICITA' COSTA  £ ",C
55 PRINT"PER UNITA' VENDUTA"
57 GET A$:IF A$="" THEN 57
58 FOR P=1 TO 10:PRINT:NEXT
60 CLR
70 GOTO 10
```

CONTAPEZZI * *

Questo programma vi consente di contare un certo quantitativo di merce. Semplicemente digitando sulla tastiera un numero alla volta. Non è neppure necessario premere RETURN il computer visualizza immediatamente il valore totale. I tasti utilizzabili sono quelli compresi tra 0 e 9.

```
10 PRINT"[CLEAR]"
20 GOSUB 200
30 GET A$:IF A$="" THEN 30
35 PRINT
50 IF A$="C" THEN BNT=0:GOTO 90
60 IF ASC(A$)<48 OR ASC(A$)>57 THEN 30
70 N=VAL(A$)
80 T=T+N
90 GOSUB 200
100 PRINT"TOTALE":T
110 GOTO 30
200 PRINT"[CLEAR]"
210 PRINT"          SOMMATORE"
220 FOR L=1 TO 22:PRINT"●":NEXT
230 RETURN
```

INVENTARIO * *

Il seguente programma è stato preparato per eseguire l'inventario di 10 categorie di prodotti.

Le categorie dei prodotti sono individuati dai numeri da 1 a 10.

Si possono introdurre un numero indeterminato di elementi per ogni categoria, senza con questo dover rispettare alcun ordine particolare.

Il valore 0 segnerà al computer che l'introduzione dei dati è terminata e provocherà la visualizzazione del gran totale relativo ad ogni categoria.

La linea 25 richiede il numero della categoria

se 0 salta alla linea 70 per il gran totale, diversamente procede alla linea 40.

L'input del numero di categoria è controllato per evitare l'acquisizione di dati non validi.

La linea 50 chiede il quantitativo di merce per categoria.

La linea 60 attua la diramazione alle diverse categorie.

Il programma ricicla continuamente fino alla digitazione del valore 0.

```
20 PRINT"[CLEAR]":REM  PULIZIA SCHERMO
25 INPUT "CODICE MERCE":C
30 IF C=0 THEN 70
40 IF C>10 THEN 20
50 INPUT "QUANTITA'":Q
60 ON CGOTO 201,202,203,204,205,206,20
  7,208,209,210
```

```

70 PRINT"[CLEAR]"
90 GOSUB 250
100 PRINT:PRINT:PRINT
110 INPUT "PREMI RETURN PER CONTINUARE"
    :K$
120 CLR :GOTO 20
201 J=J+Q:GOTO 20
202 K=K+Q:GOTO 20
203 L=L+Q:GOTO 20
204 M=M+Q:GOTO 20
205 N=N+Q:GOTO 20
206 R=R+Q:GOTO 20
207 S=S+Q:GOTO 20
208 T=T+Q:GOTO 20
209 U=U+Q:GOTO 20
210 V=V+Q:GOTO 20
220 END
250 PRINT"MERCI"
260 PRINT"CODICE", "QUANTITA'"
270 PRINT
300 PRINT"1",J
310 PRINT"2",K
320 PRINT"3",L
330 PRINT"4",M
340 PRINT"5",N
350 PRINT"6",R
360 PRINT"7",S
370 PRINT"8",T
380 PRINT"9",U
390 PRINT"10",V
400 RETURN

```

ORE E SALARI * *

Queste utili linee calcolano il numero totale di ore lavorative a tempo pieno e il numero di ore di lavoro straordinario.

Viene fornito anche l'ammontare del pagamento finale arrotondato all'unità.

Il programma sa che lo straordinario inizia dopo le 40 ore.

```
10 PRINT"[CLEAR]"
20 PRINT"STIPENDIO PER UN' ORA"
30 INPUT "£ ":P
40 PRINT"NUMERO DI ORE DA      RETRIBU
   IRE"
50 INPUT H
60 IF H>40 THEN OT=H-40:GOTO 100
70 W=H*P
80 PRINT"STIPENDIO LORDO"
90 PRINT"£":W
100 END
110 W=(40*P)+(OT*P*1.5)
120 GOTO 80
```

AL PENNY PIU' VICINO

★ ★

Questo programma è utile per arrotondare un numero contenente centesimi, al decimale più prossimo.

Per esempio, da 151,6972 a 151,70.

Questo piccolo programma può costituire una utile sottoprocedura in un programma più vasto.

Per fare ciò richiamatelo con un opportuno comando GOSUB e modificate i numeri di linea.


In questo caso non dimenticate di cancellare il contenuto della linea 180 e di sostituirlo con un RETURN.

```
10 PRINT"[CLEAR]"
20 PRINT"INTRODICI UN NUMERO    CON PIU
  / DI DUE CIFRE DECIMALI"
55 INPUT N
60 R=INT(100*N+0.5)/100
70 PRINT
80 N$="$"+RIGHT$(STR$(N),(LEN(STR$(N)
  )-1))
90 R$="$"+RIGHT$(STR$(R),(LEN(STR$(R)
  )-1))
100 :PRINT"[CLEAR]"
104 PRINTN$
110 PRINT" ARROTONDATO DIVENTA:"
115 PRINTR$
```

```

120 PRINT:PRINT:PRINT
130 PRINT"PREMI UN TASTO PER      CONTINU
    ARE"
160 GET A$:IF A$="" THEN 160
180 CLR :GOTO 10

```



AUMENTO DI VALORE MONETARIO

★ ★

Questa sezione del libro include una serie di programmi relativi alla gestione della contabilità domestica e spicciola.

Questo primo programma mostra come cresce il valore di un investimento ad un certo tasso di interesse annuo, calcolato mensilmente.

Il primo valore da inserire è l'ammontare iniziale dell'investimento.

Successivamente viene richiesto il tasso di interesse annuo e il numero di mesi da visualizzare.

La linea 10 cancella lo schermo

le linee dalla 20 alla 40 acquisiscono i dati, mentre dalla 70 alla 100 vengono prodotti i risultati.

La linea 85 genera un rallentamento della visualizzazione per permettere una più comoda lettura dei dati.



```
10 PRINT"[CLEAR]": REM  PULIZIA SCHERM
   0
20 PRINT"CAPITALE £":INPUT P
30 PRINT"INTERESSE ANNUO %":INPUT R
40 PRINT"NUMERO DEI MESI":INPUT M
50 PRINT"[CLEAR]": REM  PULIZIA SCHERM
   0
60 PRINT"MESE", "BILANCIO"
```



```

70 FOR Q=1 TO M
80 I=(P*(0.01*R))/12
90 P=P+I
100 PP=INT(100*P+0.5)/100
110 PRINTQ,PP
120 FOR T=1 TO 400:NEXTT
130 NEXTQ
140 PRINT:PRINT
150 INPUT "PER PROSEQUIRE PREMI RETURN
";K$
160 CLR :GOTO 10

```



PAGAMENTO DI AUTOMOBILI

★ ★

Dovete acquistare una nuova auto?

Servitevi del vostro computer per valutare velocemente il costo mensile di ogni modello.

Immaginate di volere una vettura da 8 milioni e di poterne anticipare solo uno.

Decidete quindi di rateizzare il rimanente importo in 36 mesi.

L'interesse annuale sulle automobili è del 15%.

Impostate i dati e otterrete il valore dell'ammontare di ogni rata.

```
10 PRINT"CLEAR": REM  PULIZIA SCHERM
   0
20 GOSUB 250
25 PRINT
30 PRINT"      PAGAMENTO RATE      A
   UTOMOBILE"
40 GOSUB 250
50 PRINT
60 PRINT" PREZZO DI ACQUISTO £"
70 INPUT T
80 PRINT" AMMONTARE DELLA      SOMMA
   RIMASTA £"
```

```

90 INPUT R
100 PRINT"    NUMERO DEI MESI"
110 INPUT N
120 PRINT"    INTERESSE ANNUO %"
130 INPUT I
150  $P=(T-R)*I/(1-1/(1+I)^N)$ 
160 PP=INT(100*P+0.5)/100
170 PRINT
180 PRINT"PAGAMENTO £",PP
190 PRINT
200 GOSUB 250
210 PRINT
220 INPUT "PER PROSEGUIRE PREMI RETURN
        " :K$
230 CLR
240 GOTO 10
250 PRINT" ";FOR L=1 TO 20
260 PRINT "●";
270 NEXTL
330 RETURN

```

GENERATORE DI CARATTERI *

Questo programma vi consentirà di riprogrammare i caratteri desiderati sul vostro VIC-20. I cursori vi consentono di muovervi in una delle 4 direzioni disponibili, posizionandovi sull'elemento che volete mutare. Tutte le successive fasi del procedimento sono sufficientemente autoesplicative.

```
100 POKE 36879,27:PRINT"[CLEAR]GENERATO  
RE CARATTERI"  
110 PRINT"[2 DOWN] POCKET GROUP"  
140 X=PEEK(56)-2:POKE 52,X:POKE 56,X:PO  
KE 51,PEEK(55):CLR  
150 CS=256*PEEK(52)+PEEK(51)  
160 FOR I=CS TO CS+511:POKE I,PEEK(I+32  
768-CS):NEXT  
170 S=7680:CL=22  
180 CR=0:LN=200:P=12:BG=3:BR=1  
190 POKE 36879,BG*16+BR  
200 DEF FNA(XX)=S+R*CL+C:REM VIDEO PO  
KE LOC  
210 DEF FNB(XX)=8*R+C:REM VIDEO POKE  
CHAR
```

```

220 GOTO 580
230 PRINT"[CLEAR]":GOSUB 810
240 PRINT"[HOME]";:FOR I=0 TO 7:PRINT".
.....":NEXT F=0
250 PRINT"[HOME]":R=0:C=0
260 Z=FNA(0)
270 IF F=0 THEN POKE Z,PEEK(Z)+128:GOTO
    310
280 IF Z=ZL THEN 300
290 POKE ZL,IL:POKE ZL+30720,BC:ZL=Z:IL
    =PEEK(ZL)
300 POKE Z,32:POKE Z+30720,2
310 GET A$:IF A$="" THEN 310
320 IF F=0 THEN POKE Z,PEEK(Z)-128
330 REM CONTROLLO CURSORE
340 IF A$="Q" THEN POKE 56,PEEK(56)+2:P
    OKE 36869,240:PRINT"[CLEAR]":END
350 IF A$="[RIGHT]" AND C=7 THEN C=0:GO
    TO 260
360 IF A$="[RIGHT]" THEN C=C+1:GOTO 260
370 IF A$="[LEFT]" AND C=0 THEN C=7:GOT
    O 260
380 IF A$="[LEFT]" THEN C=C-1:GOTO 260
390 IF A$="[DOWN]" AND R=7 THEN R=0:GOT
    O 260
400 IF A$="[DOWN]" THEN R=R+1:GOTO 260
410 IF A$="[UP]" AND R=0 THEN R=7:GOTO
    260
420 IF A$="[UP]" THEN R=R-1:GOTO 260
430 IF A$="[HOME]" THEN 250
440 IF F=1 THEN 540
450 REM DEFINISCE CARATTERE
460 IF A$="+" THEN POKE Z,81:GOTO 260
470 IF A$="-" THEN POKE Z,46:GOTO 260

```

```

480 IF A$="=" THEN 680
490 IF A$="[CLEAR]" THEN 240
500 IF A$="R" THEN 580
510 IF A$="B" THEN 770
520 GOTO 260
530 REM  OPTIONE  CARATTERE
540 CR=FNB(0)
550 IF A$="N" THEN POKE 36869,240:GOTO
    230
560 IF A$="E" THEN POKE 36869,240:F=0:G
    OTO 730
570 GOTO 260
580 POKE 36869,255:R=4:C=0:ZL=FNA(0):IL
    =32
590 PRINT"[CLEAR]@ABCDEFGH":PRINT"HIJKLM
    NO":PRINT"PQRSTUVWXYZ":PRINT"XYZI£]↑"&
    :F=1
600 PRINT" !"+CHR$(34)+"#%&'":PRINT"()
    *+,-./":PRINT"01234567":PRINT"89:;<
    =>?"
610 PRINT"[HOME]SPC(12);"[RVS]OPTIONE[
    RVS]":PRINT
620 PRINTSPC(10);"[RVS]IN NUOVO CARATTER
    E[RVOFF]"
630 PRINTSPC(10);"[RVS]E EDIT  CARATTER
    E[RVOFF]"
640 PRINTSPC(10);"[RVS]O FINE
    [RVOFF]"
650 BC=PEEK(38400)
660 GOTO 260
670 REM  UPDATE  TABELLA  CHAR
680 PRINT"[HOME]";X=CS+8*CR:FOR R=0 TO
    7:SM=0:FOR C=0 TO 7:D=7-C
690 SM=SM-2↑D*(PEEK(FNA(0))=81):NEXT

```

```

700 POKE X+R,SM:PRINTSPC(3);SM:NEXT
710 R=0:C=0:GOTO 260
720 REM EDIT CHAR IN TABELLA
730 X=CS+8*CR:PRINT"[CLEAR]":FOR R=0 TO
    7:Y=PEEK(X+R):FOR C=0 TO 7:Z=FNA(0
    )
740 Q=46:Y=Y*2:IF Y>255 THEN Q=81:Y=Y-2
    56
750 POKE Z,Q:NEXTC,R:R=0:C=0:GOSUB 810:
    GOTO 260
760 REM DEFINE CHAR
770 X=CS+8*CR:PRINT"[HOME][8 DOWN]"
780 PRINTLN:"READ X: FOR I=X TO X+7: RE
    AD Y: POKE X,Y: NEXT":LN=LN+10
790 PRINTLN:"DATA":X):FOR I=X TO X+7:PR
    INT"[LEFT]":PEEK(I):NEXTI:PRINT
800 GOTO 260
810 PRINT"[HOME]";SPC(13)"[RVS]OPTION[R
    VOFF]":PRINT
820 PRINTSPC(P);"[RVS]+[RVOFF] ADD DOT"
830 PRINTSPC(P);"[RVS]-[RVOFF] CANCEL"
840 PRINTSPC(P);"[RVS]=[RVOFF] UPDATE"
850 PRINTSPC(P);"[RVS]B[RVOFF] BASIC "
860 PRINTSPC(P);"[RVS]R[RVOFF] REVISI"
870 PRINTSPC(P);"[RVS]Q[RVOFF] FINE "
880 RETURN

```

EFFETTI SONORI *

Il VIC—20 può essere un interessante generatore di effetti sonori.

In questo programma ne vengono presentati alcuni tra i più interessanti: piano, computer mania, esplosione, bomba, allarme rosso, laser, sirena, busy busy, telefono, uccello, vittoria, onde dell'oceano, ufo, ufo in atterraggio, wolf, tick tock, door opening, blips.

```
1 PRINT"[CLEAR]1 PIANO
2 PRINT"2 COMPUTER MANIA
3 PRINT"3 ESPLOSIONE
4 PRINT"4 BOMBA
5 PRINT"5 ALLARME ROSSO
6 PRINT"6 LASER
7 PRINT"7 SIRENA
8 PRINT"8 BUSY BUSY
9 PRINT"9 TELEFONO
10 PRINT"A UCCELLO
11 PRINT"B VITTORIA
12 PRINT"C ONDE OCEANO
13 PRINT"D UFO CHE VA'
14 PRINT"E UFO ATTERRAGGIO
15 PRINT"F UFO (UFFA)
16 PRINT"G WOLF
17 PRINT"H GIRO
18 PRINT"I TICK TOCK
```



```

19 PRINT"J DOOR OPENING
20 PRINT"K BLIPS
50 A4=36874:A5=A4+1:A6=A5+1:A7=A6+1:A8
  =A7+1
100 GET A$:IF A$="" THEN 100
110 PRINT"[HOME][DOWN]" TAB(20)A$
120 A=ASC(A$)-55:IF A>9 THEN 310
200 A=VAL(A$):IF A=0 THEN 100
300 ON AGOSUB 1100,1200,1300,1400,1500,
  1600,1700,1800,1900:GOTO 100
310 ON A-9GOSUB 2000,2100,2200,2300,240
  0,2500,2600,2700,2800,2900,3000
400 GOTO 100
1100 POKE A8,15:FOR L=250 TO 200 STEP -2
  :POKE A6,L:FOR M=1 TO 100:NEXT:NEXT
1110 FOR L=205 TO 250 STEP 2:POKE A6,L:F
  OR M=1 TO 100:NEXT:NEXT:GOTO 10000
1200 POKE A8,15:FOR L=1 TO 100:POKE A6,I
  NT(RND(1)*128)+128:FOR M=1 TO 10:NE
  XT:NEXT:GOTO 10000
1300 POKE A7,220:FOR L=15 TO 0 STEP -1:P
  OKE A8,L:FOR M=1 TO 100:NEXT:NEXT:G
  OTO 10000
1400 POKE A8,10:FOR L=230 TO 128 STEP -1
  :POKE A6,L:FOR M=1 TO 20:NEXT:NEXT:
  POKE A6,0:POKE A7,200
1410 FOR L=15 TO 0 STEP -.05:POKE A8,L:N
  EXT:GOTO 10000
1500 POKE A8,15:FOR L=1 TO 10:FOR M=180
  TO 235 STEP 2:POKE A6,M:FOR N=1 TO
  10:NEXT:NEXT
1510 POKE A6,0:FOR M=1 TO 100:NEXT:NEXT:
  GOTO 10000
1600 POKE A8,15:FOR L=1 TO 30:FOR M=250

```

```

      TO 240 STEP -1:POKE A6,M:NEXT
1610 FOR M=240 TO 250:POKE A6,M:NEXT:POKE
      A6,0:NEXT:GOTO 10000
1700 POKE A8,15:FOR L=1 TO 10:POKE A5,20
      0:FOR M=1 TO 500:NEXT
1710 POKE A5,0:POKE A6,200:FOR M=1 TO 50
      0:NEXT:POKE A6,0:NEXT:GOTO 10000
1800 POKE A8,15:FOR L=1 TO 15:POKE A6,16
      0:FOR M=1 TO 400:NEXT
1810 POKE A6,0:FOR M=1 TO 400:NEXT:NEXT:
      GOTO 10000
1900 POKE A8,15:FOR L=1 TO 5:FOR M=1 TO
      50:POKE A6,220:FOR N=1 TO 5:NEXT
1910 POKE A6,0:NEXT:FOR M=1 TO 3000:NEXT
      :NEXT:GOTO 10000
2000 POKE A8,15:FOR L=1 TO 20:FOR M=254
      TO 240+INT(RND(1)*10) STEP -1:POKE
      A6,M:NEXT
2010 POKE A6,0:FOR M=1 TO INT(RND(1)*100
      )+120:NEXT:NEXT:GOTO 10000
2100 POKE A8,15:POKE A4,170:POKE A7,240:
      FOR L=1 TO 2000:NEXT:GOTO 10000
2200 POKE A7,180:FOR L=1 TO 10:D=INT(RND
      (1)*5)*50+50:FOR M=3 TO 15:POKE A8,
      M
2210 FOR N=1 TO D:NEXT:NEXT:FOR M=15 TO
      3 STEP -1:POKE A8,M:FOR N=1 TO D:NE
      XT:NEXT:NEXT:GOTO 10000
2300 POKE A8,15:FOR L=130 TO 254:POKE A6
      ,L:FOR M=1 TO 40:NEXT:NEXT:GOTO 100
      00
2400 POKE A8,15:FOR L=1 TO 20:FOR M=220-
      L TO 160-L STEP -4:POKE A6,M:NEXT
2410 FOR M=160-L TO 220-L STEP 4:POKE A6

```

```

,M: NEXT: NEXT: GOTO 10000
2500 POKE A8,15: FOR L=1 TO 15: FOR M=200
    TO 220+L*2: POKE A6,M: NEXT: NEXT: GOTO
    10000
2600 POKE A8,15: FOR L=148 TO 220 STEP .7
    : POKE A6,L: NEXT: FOR L=128 TO 200
2610 POKE A6,L: NEXT: FOR L=200 TO 128 STE
    P -.1: POKE A6,L: NEXT: GOTO 10000
2700 POKE A8,15: FOR L=1 TO 10: POKE A4,20
    0: FOR M=1 TO 10: NEXT: POKE A4,0: FOR
    M=1 TO 100: NEXT: NEXT: GOTO 10000
2800 POKE A8,15: FOR L=1 TO 10: POKE A5,20
    0: FOR M=1 TO 10: NEXT: POKE A5,0
2810 FOR M=1 TO 300: NEXT: POKE A4,200: FOR
    M=1 TO 10: NEXT: POKE A4,0: FOR M=1 T
    O 300: NEXT: NEXT: GOTO 10000
2900 POKE A8,15: B=0: FOR L=128 TO 255 STE
    P .11: POKE A4,L: FOR M=1 TO 10: NEXT: B
    =B+1
2910 IF B=3 THEN B=0: POKE A4,0
2920 NEXT: GOTO 10000
3000 POKE A8,15: POKE A6,220: FOR L=1 TO 5
    : NEXT: POKE A6,0: FOR L=1 TO 500: NEXT
    : POKE A6,200
3010 FOR L=1 TO 5: NEXT: POKE A6,0: FOR L=1
    TO 500: NEXT: POKE A8,0: GOTO 10000
10000 POKE A4,0: POKE A6,0: POKE A7,0: POKE
    A8,0: RETURN

```

LIBRI DI SYSTEM -- libro mensile di Commodore Computer Club - Edizioni System Editoriale
s.r.l. - v.le Famagosta, 75 - 20142 Milano - Tel. 02/8467348 - Dir. responsabile: Michele di Pisa -
Reg. Trib. di Milano n. 370/82 - Sped. Abb. post. gr. III/70 - Stampa: Lito 3 - Cologno Monzese.



Lire 6.500
Anno 2 N. 3 —Aprile 1985
Distr. MePe